

# Defining (Co)datatypes and Primitively (Co)recursive Functions in Isabelle/HOL

Julian Biendarra, Jasmin Blanchette,  
Martin Desharnais, Lorenz Panny,  
Andrei Popescu, and Dmitriy Traytel

23 May 2024

## Abstract

This tutorial describes the definitional package for datatypes and co-datatypes, and for primitively recursive and corecursive functions, in Isabelle/HOL. The following commands are provided: **datatype**, **datatype\_compat**, **primrec**, **codatatype**, **primcorec**, **primcorecursive**, **bnf**, **lift\_bnf**, **copy\_bnf**, **bnf\_axiomatization**, **print\_bnfs**, and **free\_constructors**.

## Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
<b>2</b>	<b>Defining Datatypes</b>	<b>5</b>
2.1	Introductory Examples . . . . .	5
2.1.1	Nonrecursive Types . . . . .	5
2.1.2	Simple Recursion . . . . .	6
2.1.3	Mutual Recursion . . . . .	6
2.1.4	Nested Recursion . . . . .	7
2.1.5	Auxiliary Constants . . . . .	7
2.2	Command Syntax . . . . .	9
2.2.1	<b>datatype</b> . . . . .	9
2.2.2	<b>datatype_compat</b> . . . . .	12
2.3	Generated Constants . . . . .	13
2.4	Generated Theorems . . . . .	13
2.4.1	Free Constructor Theorems . . . . .	14

2.4.2	Functorial Theorems . . . . .	16
2.4.3	Inductive Theorems . . . . .	21
2.5	Proof Method . . . . .	22
2.5.1	<i>countable_datatype</i> . . . . .	22
2.6	Antiquotation . . . . .	22
2.6.1	<i>datatype</i> . . . . .	22
2.7	Compatibility Issues . . . . .	22
<b>3</b>	<b>Defining Primitively Recursive Functions</b>	<b>24</b>
3.1	Introductory Examples . . . . .	24
3.1.1	Nonrecursive Types . . . . .	24
3.1.2	Simple Recursion . . . . .	25
3.1.3	Mutual Recursion . . . . .	25
3.1.4	Nested Recursion . . . . .	26
3.1.5	Nested-as-Mutual Recursion . . . . .	27
3.2	Command Syntax . . . . .	28
3.2.1	<b>primrec</b> . . . . .	28
3.3	Generated Theorems . . . . .	29
3.4	Recursive Default Values for Selectors . . . . .	30
3.5	Compatibility Issues . . . . .	31
<b>4</b>	<b>Defining Codatatypes</b>	<b>31</b>
4.1	Introductory Examples . . . . .	31
4.1.1	Simple Corecursion . . . . .	31
4.1.2	Mutual Corecursion . . . . .	32
4.1.3	Nested Corecursion . . . . .	33
4.2	Command Syntax . . . . .	33
4.2.1	<b>codatatype</b> . . . . .	33
4.3	Generated Constants . . . . .	33
4.4	Generated Theorems . . . . .	33
4.4.1	Coinductive Theorems . . . . .	34
4.5	Antiquotation . . . . .	35
4.5.1	<i>codatatype</i> . . . . .	35
<b>5</b>	<b>Defining Primitively Corecursive Functions</b>	<b>36</b>
5.1	Introductory Examples . . . . .	37
5.1.1	Simple Corecursion . . . . .	37
5.1.2	Mutual Corecursion . . . . .	38
5.1.3	Nested Corecursion . . . . .	39
5.1.4	Nested-as-Mutual Corecursion . . . . .	40
5.1.5	Constructor View . . . . .	41

1	INTRODUCTION	3
	5.1.6	Destructor View . . . . . 42
5.2	Command Syntax . . . . . 43	
	5.2.1	<b>primcorec</b> and <b>primcorecursive</b> . . . . . 43
5.3	Generated Theorems . . . . . 45	
<b>6</b>	<b>Registering Bounded Natural Functors</b>	<b>46</b>
6.1	Bounded Natural Functors . . . . . 46	
6.2	Introductory Examples . . . . . 47	
6.3	Command Syntax . . . . . 51	
	6.3.1	<b>bnf</b> . . . . . 51
	6.3.2	<b>lift_bnf</b> . . . . . 52
	6.3.3	<b>copy_bnf</b> . . . . . 54
	6.3.4	<b>bnf_axiomatization</b> . . . . . 54
	6.3.5	<b>print_bnfs</b> . . . . . 56
<b>7</b>	<b>Deriving Destructors and Constructor Theorems</b>	<b>56</b>
7.1	Command Syntax . . . . . 56	
	7.1.1	<b>free_constructors</b> . . . . . 56
	7.1.2	<b>simps_of_case</b> . . . . . 57
	7.1.3	<b>case_of_simps</b> . . . . . 58
<b>8</b>	<b>Selecting Plugins</b>	<b>59</b>
8.1	Code Generator . . . . . 59	
8.2	Size . . . . . 59	
8.3	Transfer . . . . . 60	
8.4	Lifting . . . . . 61	
8.5	Quickcheck . . . . . 61	
8.6	Program Extraction . . . . . 62	
<b>9</b>	<b>Known Bugs and Limitations</b>	<b>62</b>

## 1 Introduction

The 2013 edition of Isabelle introduced a definitional package for freely generated datatypes and codatatypes. This package replaces the earlier implementation due to Berghofer and Wenzel [1]. Perhaps the main advantage of the new package is that it supports recursion through a large class of non-datatypes, such as finite sets:

```
datatype 'a treefs = Nodefs (lblfs: 'a) (subfs: "'a treefs fset")
```

Another strong point is the support for local definitions:

```

context linorder
begin
datatype flag = Less | Eq | Greater
end

```

Furthermore, the package provides a lot of convenience, including automatically generated discriminators, selectors, and relators as well as a wealth of properties about them.

In addition to inductive datatypes, the package supports coinductive datatypes, or *codatatypes*, which allow infinite values. For example, the following command introduces the type of lazy lists, which comprises both finite and infinite values:

```

codatatype 'a llist = LNil | LCons 'a "'a llist"

```

Mixed inductive–coinductive recursion is possible via nesting. Compare the following four Rose tree examples:

```

datatype 'a treeff = Nodeff 'a "'a treeff list"
datatype 'a treefi = Nodefi 'a "'a treefi llist"
codatatype 'a treeif = Nodeif 'a "'a treeif list"
codatatype 'a treeii = Nodeii 'a "'a treeii llist"

```

The first two tree types allow only paths of finite length, whereas the last two allow infinite paths. Orthogonally, the nodes in the first and third types have finitely many direct subtrees, whereas those of the second and fourth may have infinite branching.

The package is part of *Main*. Additional functionality is provided by the theory `~/src/HOL/Library/BNF_Axiomatization.thy`.

The package, like its predecessor, fully adheres to the LCF philosophy [5]: The characteristic theorems associated with the specified (co)datatypes are derived rather than introduced axiomatically.<sup>1</sup> The package is described in a number of scientific papers [2, 4, 9, 11]. The central notion is that of a *bounded natural functor* (BNF)—a well-behaved type constructor for which nested (co)recursion is supported.

This tutorial is organized as follows:

- Section 2, “Defining Datatypes,” describes how to specify datatypes using the **datatype** command.
- Section 3, “Defining Primitively Recursive Functions,” describes how to specify functions using **primrec**. (A separate tutorial [6] describes the more powerful **fun** and **function** commands.)

---

<sup>1</sup>However, some of the internal constructions and most of the internal proof obligations are omitted if the *quick\_and\_dirty* option is enabled.

- Section 4, “Defining Codatatypes,” describes how to specify codatatypes using the **codatatype** command.
- Section 5, “Defining Primitively Corecursive Functions,” describes how to specify functions using the **primcorec** and **primcorecursive** commands. (A separate tutorial [3] describes the more powerful **corec** and **corecursive** commands.)
- Section 6, “Registering Bounded Natural Functors,” explains how to use the **bnf** command to register arbitrary type constructors as BNFs.
- Section 7, “Deriving Destructors and Constructor Theorems,” explains how to use the command **free\_constructors** to derive destructor constants and theorems for freely generated types, as performed internally by **datatype** and **codatatype**.
- Section 8, “Selecting Plugins,” is concerned with the package’s interoperability with other Isabelle packages and tools, such as the code generator, Transfer, Lifting, and Quickcheck.
- Section 9, “Known Bugs and Limitations,” concludes with known open issues.

Comments and bug reports concerning either the package or this tutorial should be directed to the second author at [jasmin.blanchette@gmail.com](mailto:jasmin.blanchette@gmail.com) or to the `cl-isabelle-users` mailing list.

## 2 Defining Datatypes

Datatypes can be specified using the **datatype** command.

### 2.1 Introductory Examples

Datatypes are illustrated through concrete examples featuring different flavors of recursion. More examples can be found in the directory `~/src/HOL/Datatype_Examples`.

#### 2.1.1 Nonrecursive Types

Datatypes are introduced by specifying the desired names and argument types for their constructors. *Enumeration* types are the simplest form of datatype. All their constructors are nullary:

**datatype** *trool* = *Truee* | *Faalse* | *Perhaaps*

*Truee*, *Faalse*, and *Perhaaps* have the type *tbool*.

Polymorphic types are possible, such as the following option type, modeled after its homologue from the *HOL.Option* theory:

```
datatype 'a option = None | Some 'a
```

The constructors are *None* :: 'a option and *Some* :: 'a ⇒ 'a option.

The next example has three type parameters:

```
datatype ('a, 'b, 'c) triple = Triple 'a 'b 'c
```

The constructor is *Triple* :: 'a ⇒ 'b ⇒ 'c ⇒ ('a, 'b, 'c) triple. Unlike in Standard ML, curried constructors are supported. The uncurried variant is also possible:

```
datatype ('a, 'b, 'c) triple_u = Triple_u " 'a * 'b * 'c"
```

Occurrences of nonatomic types on the right-hand side of the equal sign must be enclosed in double quotes, as is customary in Isabelle.

### 2.1.2 Simple Recursion

Natural numbers are the simplest example of a recursive type:

```
datatype nat = Zero | Succ nat
```

Lists were shown in the introduction. Terminated lists are a variant that stores a value of type 'b at the very end:

```
datatype ('a, 'b) tlist = TNil 'b | TCons 'a "( 'a, 'b) tlist"
```

### 2.1.3 Mutual Recursion

*Mutually recursive* types are introduced simultaneously and may refer to each other. The example below introduces a pair of types for even and odd natural numbers:

```
datatype even_nat = Even_Zero | Even_Succ odd_nat
and odd_nat = Odd_Succ even_nat
```

Arithmetic expressions are defined via terms, terms via factors, and factors via expressions:

```
datatype ('a, 'b) exp =
  Term "( 'a, 'b) trm" | Sum "( 'a, 'b) trm" "( 'a, 'b) exp"
and ('a, 'b) trm =
  Factor "( 'a, 'b) fct" | Prod "( 'a, 'b) fct" "( 'a, 'b) trm"
and ('a, 'b) fct =
  Const 'a | Var 'b | Expr "( 'a, 'b) exp"
```

### 2.1.4 Nested Recursion

*Nested recursion* occurs when recursive occurrences of a type appear under a type constructor. The introduction showed some examples of trees with nesting through lists. A more complex example, that reuses our *option* type, follows:

```
datatype 'a btree =
  BNode 'a "'a btree option" "'a btree option"
```

Not all nestings are admissible. For example, this command will fail:

```
datatype 'a wrong = W1 | W2 "'a wrong  $\Rightarrow$  'a"
```

The issue is that the function arrow  $\Rightarrow$  allows recursion only through its right-hand side. This issue is inherited by polymorphic datatypes defined in terms of  $\Rightarrow$ :

```
datatype ('a, 'b) fun_copy = Fun "'a  $\Rightarrow$  'b"
datatype 'a also_wrong = W1 | W2 "('a also_wrong, 'a) fun_copy"
```

The following definition of *'a*-branching trees is legal:

```
datatype 'a ftree = FTLeaf 'a | FTNode "'a  $\Rightarrow$  'a ftree"
```

And so is the definition of hereditarily finite sets:

```
datatype hfset = HFSet "hfset fset"
```

In general, type constructors  $(a_1, \dots, a_m) t$  allow recursion on a subset of their type arguments  $a_1, \dots, a_m$ . These type arguments are called *live*; the remaining type arguments are called *dead*. In  $'a \Rightarrow 'b$  and  $(a, 'b) fun\_copy$ , the type variable *'a* is dead and *'b* is live.

Type constructors must be registered as BNFs to have live arguments. This is done automatically for datatypes and codatatypes introduced by the **datatype** and **codatatype** commands. Section 6 explains how to register arbitrary type constructors as BNFs.

Here is another example that fails:

```
datatype 'a pow_list = PNil 'a | PCons "('a * 'a) pow_list"
```

This attempted definition features a different flavor of nesting, where the recursive call in the type specification occurs around (rather than inside) another type constructor.

### 2.1.5 Auxiliary Constants

The **datatype** command introduces various constants in addition to the constructors. With each datatype are associated set functions, a map function, a

predicator, a relator, discriminators, and selectors, all of which can be given custom names. In the example below, the familiar names *null*, *hd*, *tl*, *set*, *map*, and *list\_all2* override the default names *is\_Nil*, *un\_Cons1*, *un\_Cons2*, *set\_list*, *map\_list*, and *rel\_list*:

```
datatype (set: 'a) list =
  null: Nil
| Cons (hd: 'a) (tl: "'a list")
for
  map: map
  rel: list_all2
  pred: list_all
where
  "tl Nil = Nil"
```

The types of the constants that appear in the specification are listed below.

```
Constructors:  Nil :: 'a list
               Cons :: 'a ⇒ 'a list ⇒ 'a list
Discriminator: null :: 'a list ⇒ bool
Selectors:     hd  :: 'a list ⇒ 'a
               tl  :: 'a list ⇒ 'a list
Set function:  set :: 'a list ⇒ 'a set
Map function:  map :: ('a ⇒ 'b) ⇒ 'a list ⇒ 'b list
Relator:       list_all2 :: ('a ⇒ 'b ⇒ bool) ⇒ 'a list ⇒ 'b list ⇒ bool
```

The discriminator *null* and the selectors *hd* and *tl* are characterized by the following conditional equations:

$$\text{null } xs \implies xs = \text{Nil} \quad \neg \text{null } xs \implies \text{Cons } (\text{hd } xs) (\text{tl } xs) = xs$$

For two-constructor datatypes, a single discriminator constant is sufficient. The discriminator associated with *Cons* is simply  $\lambda xs. \neg \text{null } xs$ .

The **where** clause at the end of the command specifies a default value for selectors applied to constructors on which they are not a priori specified. In the example, it is used to ensure that the tail of the empty list is itself (instead of being left unspecified).

Because *Nil* is nullary, it is also possible to use  $\lambda xs. xs = \text{Nil}$  as a discriminator. This is the default behavior if we omit the identifier *null* and the associated colon. Some users argue against this, because the mixture of constructors and selectors in the characteristic theorems can lead Isabelle’s automation to switch between the constructor and the destructor view in surprising ways.

The usual mixfix syntax annotations are available for both types and constructors. For example:



```
datatype ('a, 'b) prod (infixr "*" 20) = Pair 'a 'b
```

```
datatype (set: 'a) list =
  null: Nil ("[]")
| Cons (hd: 'a) (tl: "'a list") (infixr "#" 65)
for
  map: map
  rel: list_all2
  pred: list_all
```

Incidentally, this is how the traditional syntax can be set up:

```
syntax "_list" :: "args => 'a list" ("[(_)"])
```

**translations**


```
"[x, xs]" == "x # [xs]"
```

```
"[x]" == "x # []"
```


## 2.2 Command Syntax

### 2.2.1 datatype


```
datatype : local_theory → local_theory
```





*dt-options*



*plugins*



*dt-spec**map-rel-pred*

The **datatype** command introduces a set of mutually recursive datatypes specified by their constructors.

The syntactic entity *target* can be used to specify a local context (e.g., (*in linorder*) [12]), and *prop* denotes a HOL proposition.


The optional *target* is optionally followed by a combination of the following options:

- The *plugins* option indicates which plugins should be enabled (*only*) or disabled (*del*). By default, all plugins are enabled.
- The *discs\_sels* option indicates that discriminators and selectors should be generated. The option is implicitly enabled if names are specified for discriminators or selectors.


The optional **where** clause specifies default values for selectors. Each proposition must be an equation of the form  $un\_D (C \dots) = \dots$ , where  $C$  is a constructor and  $un\_D$  is a selector.

The left-hand sides of the datatype equations specify the name of the type to define, its type parameters, and additional information:

*dt-name*



*tyargs*




The syntactic entity *name* denotes an identifier, *mixfix* denotes the usual parenthesized mixfix notation, and *typefree* denotes fixed type variable (*'a*, *'b*, ...) [12].

The optional names preceding the type variables allow to override the default names of the set functions ( $set_{1\_t}, \dots, set_{m\_t}$ ). Type arguments can be marked as dead by entering *dead* in front of the type variable (e.g., (*dead 'a*)); otherwise, they are live or dead (and a set function is generated or not) depending on where they occur in the right-hand sides of the definition. Declaring a type argument as dead can speed up the type definition but will prevent any later (co)recursion through that type argument.


Inside a mutually recursive specification, all defined datatypes must mention exactly the same type variables in the same order.

*dt-ctor*



The main constituents of a constructor specification are the name of the constructor and the list of its argument types. An optional discriminator name can be supplied at the front. If discriminators are enabled (cf. the *discs\_sels* option) but no name is supplied, the default is  $\lambda x. x = C_j$  for nullary constructors and  $t.is\_C_j$  otherwise.

*dt-ctor-arg*




The syntactic entity *type* denotes a HOL type [12].

In addition to the type of a constructor argument, it is possible to specify a name for the corresponding selector. The same selector name can be reused for arguments to several constructors as long as the arguments share the same type. If selectors are enabled (cf. the *discs\_sels* option) but no name is supplied, the default name is  $un\_C_ji$ .

### 2.2.2 datatype\_compat

**datatype\_compat** : *local\_theory*  $\rightarrow$  *local\_theory*



The **datatype\_compat** command registers new-style datatypes as old-style datatypes and invokes the old-style plugins. For example:

```
datatype_compat even_nat odd_nat
```

```
ML <Old_Datatype_Data.get_info theory type_name <even_nat>>
```

The syntactic entity *name* denotes an identifier [12].

The command is sometimes useful when migrating from the old datatype package to the new one.

A few remarks concern nested recursive datatypes:

- The old-style, nested-as-mutual induction rule and recursor theorems are generated under their usual names but with “*compat\_*” prefixed (e.g., *compat\_tree.induct*, *compat\_tree.inducts*, and *compat\_tree.rec*). These theorems should be identical to the ones generated by the old datatype package, *up to the order of the premises*—meaning that the subgoals generated by the *induct* or *induction* method may be in a different order than before.
- All types through which recursion takes place must be new-style datatypes or the function type.

### 2.3 Generated Constants

Given a datatype  $(a_1, \dots, a_m)$   $t$  with  $m$  live type variables and  $n$  constructors  $t.C_1, \dots, t.C_n$ , the following auxiliary constants are introduced:

Case combinator:	$t.case\_t$ (rendered using the familiar <i>case-of</i> syntax)
Discriminators:	$t.is\_C_1, \dots, t.is\_C_n$
Selectors:	$t.un\_C_1k_1, \dots, t.un\_C_1k_1$
	⋮
	$t.un\_C_nk_1, \dots, t.un\_C_nk_n$
Set functions:	$t.set_1\_t, \dots, t.set_m\_t$
Map function:	$t.map\_t$
Relator:	$t.rel\_t$
Recursor:	$t.rec\_t$

The discriminators and selectors are generated only if the *discs\_sels* option is enabled or if names are specified for discriminators or selectors. The set functions, map function, predicator, and relator are generated only if  $m > 0$ .

In addition, some of the plugins introduce their own constants (Section 8). The case combinator, discriminators, and selectors are collectively called *destructors*. The prefix “*t.*” is an optional component of the names and is normally hidden.

### 2.4 Generated Theorems

The characteristic theorems generated by **datatype** are grouped in three broad categories:

- The *free constructor theorems* (Section 2.4.1) are properties of the constructors and destructors that can be derived for any freely generated type. Internally, the derivation is performed by **free\_constructors**.

- The *functorial theorems* (Section 2.4.2) are properties of datatypes related to their BNF nature.
- The *inductive theorems* (Section 2.4.3) are properties of datatypes related to their inductive nature.

The full list of named theorems can be obtained by issuing the command `print_theorems` immediately after the datatype definition. This list includes theorems produced by plugins (Section 8), but normally excludes low-level theorems that reveal internal constructions. To make these accessible, add the line

```
declare [[bnf_internals]]
```

### 2.4.1 Free Constructor Theorems

The free constructor theorems are partitioned in three subgroups. The first subgroup of properties is concerned with the constructors. They are listed below for 'a list:

*t.inject* [*iff*, *induct\_simp*]:

$$(x21 \# x22 = y21 \# y22) = (x21 = y21 \wedge x22 = y22)$$

*t.distinct* [*simp*, *induct\_simp*]:

$$\begin{aligned} \square &\neq x21 \# x22 \\ x21 \# x22 &\neq \square \end{aligned}$$

*t.exhaust* [*cases t*, *case\_names C<sub>1</sub> ... C<sub>n</sub>*]:

$$\llbracket y = \square \implies P; \bigwedge x21 \ x22. y = x21 \# x22 \implies P \rrbracket \implies P$$

*t.nchotomy*:

$$\forall list. list = \square \vee (\exists x21 \ x22. list = x21 \# x22)$$

In addition, these nameless theorems are registered as safe elimination rules:

*t.distinct* [**THEN notE**, *elim!*]:

$$\begin{aligned} \square = x21 \# x22 &\implies R \\ x21 \# x22 = \square &\implies R \end{aligned}$$

The next subgroup is concerned with the case combinator:

*t.case* [*simp*, *code*]:

$$\begin{aligned} (\text{case } \square \text{ of } \square \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) &= f1 \\ (\text{case } x21 \# x22 \text{ of } \square \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) &= f2 \ x21 \ x22 \end{aligned}$$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

**t.case\_cong** [*fundef\_cong*]:

$$\llbracket list = list'; list' = [] \implies f1 = g1; \bigwedge x21\ x22. list' = x21 \# x22 \implies f2\ x21\ x22 = g2\ x21\ x22 \rrbracket \implies (case\ list\ of\ [] \Rightarrow f1 \mid x21 \# x22 \Rightarrow f2\ x21\ x22) = (case\ list'\ of\ [] \Rightarrow g1 \mid x21 \# x22 \Rightarrow g2\ x21\ x22)$$

**t.case\_cong\_weak** [*cong*]:

$$list = list' \implies (case\ list\ of\ [] \Rightarrow f1 \mid x \# xa \Rightarrow f2\ x\ xa) = (case\ list'\ of\ [] \Rightarrow f1 \mid x \# xa \Rightarrow f2\ x\ xa)$$

**t.case\_distrib**:

$$h (case\ list\ of\ [] \Rightarrow f1 \mid x \# xa \Rightarrow f2\ x\ xa) = (case\ list\ of\ [] \Rightarrow h\ f1 \mid x1 \# x2 \Rightarrow h (f2\ x1\ x2))$$

**t.split**:

$$P (case\ list\ of\ [] \Rightarrow f1 \mid x \# xa \Rightarrow f2\ x\ xa) = ((list = [] \longrightarrow P\ f1) \wedge (\forall x21\ x22. list = x21 \# x22 \longrightarrow P (f2\ x21\ x22)))$$

**t.split\_asm**:

$$P (case\ list\ of\ [] \Rightarrow f1 \mid x \# xa \Rightarrow f2\ x\ xa) = (\neg (list = [] \wedge \neg P (f2\ x21\ x22))) \vee (\exists x21\ x22. list = x21 \# x22 \wedge \neg P (f2\ x21\ x22))$$

**t.splits** = *split split\_asm*

The third subgroup revolves around discriminators and selectors:

**t.disc** [*simp*]:

$$\begin{aligned} & null\ [] \\ & \neg null (x21 \# x22) \end{aligned}$$

**t.discI**:

$$\begin{aligned} list = [] & \implies null\ list \\ list = x21 \# x22 & \implies \neg null\ list \end{aligned}$$

**t.sel** [*simp*, *code*]:

$$\begin{aligned} hd (x21 \# x22) & = x21 \\ tl (x21 \# x22) & = x22 \end{aligned}$$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

**t.collapse** [*simp*]:

$$\begin{aligned} null\ list & \implies list = [] \\ \neg null\ list & \implies hd\ list \# tl\ list = list \end{aligned}$$

The [*simp*] attribute is exceptionally omitted for datatypes equipped with a single nullary constructor, because a property of the form  $x = C$  is not suitable as a simplification rule.

**t.distinct\_disc** [*dest*]:

These properties are missing for '*a list*' because there is only one

proper discriminator. If the datatype had been introduced with a second discriminator called *nonnull*, they would have read as follows:

$$\begin{aligned} \text{null list} &\Longrightarrow \neg \text{nonnull list} \\ \text{nonnull list} &\Longrightarrow \neg \text{null list} \end{aligned}$$

$$\begin{aligned} t.\text{exhaust\_disc} \text{ [case\_names } C_1 \dots C_n]: \\ \llbracket \text{null list} \Longrightarrow P; \neg \text{null list} \Longrightarrow P \rrbracket \Longrightarrow P \end{aligned}$$

$$\begin{aligned} t.\text{exhaust\_sel} \text{ [case\_names } C_1 \dots C_n]: \\ \llbracket \text{list} = [] \Longrightarrow P; \text{list} = \text{hd list} \# \text{tl list} \Longrightarrow P \rrbracket \Longrightarrow P \end{aligned}$$

$$\begin{aligned} t.\text{expand}: \\ \llbracket \text{null list} = \text{null list}'; \llbracket \neg \text{null list}; \neg \text{null list}' \rrbracket \Longrightarrow \text{hd list} = \text{hd list}' \\ \wedge \text{tl list} = \text{tl list}' \rrbracket \Longrightarrow \text{list} = \text{list}' \end{aligned}$$

$$\begin{aligned} t.\text{split\_sel}: \\ P (\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = ((\text{list} = [] \longrightarrow P \ f1) \\ \wedge (\text{list} = \text{hd list} \# \text{tl list} \longrightarrow P \ (f2 \ (\text{hd list}) \ (\text{tl list})))) \end{aligned}$$

$$\begin{aligned} t.\text{split\_sel\_asm}: \\ P (\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\neg (\text{list} = [] \wedge \neg P \\ f1 \vee \text{list} = \text{hd list} \# \text{tl list} \wedge \neg P \ (f2 \ (\text{hd list}) \ (\text{tl list})))) \end{aligned}$$

$$t.\text{split\_sels} = \text{split\_sel split\_sel\_asm}$$

$$\begin{aligned} t.\text{case\_eq\_if}: \\ (\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\text{if null list then } f1 \ \text{else} \\ f2 \ (\text{hd list}) \ (\text{tl list})) \end{aligned}$$

$$\begin{aligned} t.\text{disc\_eq\_case}: \\ \text{null list} = (\text{case list of } [] \Rightarrow \text{True} \mid uu\_ \# uua\_ \Rightarrow \text{False}) \\ (\neg \text{null list}) = (\text{case list of } [] \Rightarrow \text{False} \mid uu\_ \# uua\_ \Rightarrow \text{True}) \end{aligned}$$

In addition, equational versions of *t.disc* are registered with the *[code]* attribute. The *[code]* attribute is set by the *code* plugin (Section 8.1).

### 2.4.2 Functorial Theorems

The functorial theorems are generated for type constructors with at least one live type argument (e.g., *'a list*). They are partitioned in two subgroups. The first subgroup consists of properties involving the constructors or the destructors and either a set function, the map function, the predicator, or the relator:

$$\begin{aligned} t.\text{case\_transfer} \text{ [transfer\_rule]:} \\ \text{rel\_fun } S \ (\text{rel\_fun} \ (\text{rel\_fun } R \ (\text{rel\_fun} \ (\text{list\_all2 } R) \ S)) \ (\text{rel\_fun} \\ (\text{list\_all2 } R) \ S)) \ \text{case\_list case\_list} \end{aligned}$$

This property is generated by the *transfer* plugin (Section 8.3).



***t.sel\_transfer*** [*transfer\_rule*]:

This property is missing for 'a list because there is no common selector to all constructors.

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3).

***t.ctr\_transfer*** [*transfer\_rule*]:

$list\_all2\ R\ []\ []$

$rel\_fun\ R\ (rel\_fun\ (list\_all2\ R)\ (list\_all2\ R))\ (\#)\ (\#)$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3)

.

***t.disc\_transfer*** [*transfer\_rule*]:

$rel\_fun\ (list\_all2\ R)\ (=)\ null\ null$

$rel\_fun\ (list\_all2\ R)\ (=)\ (\lambda list.\ \neg\ null\ list)\ (\lambda list.\ \neg\ null\ list)$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3).

***t.set*** [*simp*, *code*]:

$set\ [] = \{\}$

$set\ (x21\ \# x22) = insert\ x21\ (set\ x22)$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

***t.set\_cases*** [*consumes* 1, *cases* *set*: *set<sub>i</sub> t*]:

$\llbracket e \in set\ a; \bigwedge z2.\ a = e\ \# z2 \implies thesis; \bigwedge z1\ z2.\ \llbracket a = z1\ \# z2; e \in set\ z2 \rrbracket \implies thesis \rrbracket \implies thesis$

***t.set\_intros***:

$x21 \in set\ (x21\ \# x22)$

$y \in set\ x22 \implies y \in set\ (x21\ \# x22)$

***t.set\_sel***:

$\neg\ null\ a \implies hd\ a \in set\ a$

$\llbracket \neg\ null\ a; x \in set\ (tl\ a) \rrbracket \implies x \in set\ a$

***t.map*** [*simp*, *code*]:

$map\ f\ [] = []$

$map\ f\ (x21\ \# x22) = f\ x21\ \# map\ f\ x22$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

***t.map\_disc\_iff*** [*simp*]:

$null\ (map\ f\ a) = null\ a$

***t.map\_sel***:

$\neg\ null\ a \implies hd\ (map\ f\ a) = f\ (hd\ a)$

$\neg\ null\ a \implies tl\ (map\ f\ a) = map\ f\ (tl\ a)$

***t.pred\_inject*** [*simp*]:

$list\_all\ P\ []$

$list\_all\ P\ (a\ \# aa) = (P\ a \wedge list\_all\ P\ aa)$

***t.rel\_inject*** [*simp*]:

$$\begin{aligned} & \text{list\_all2 } R \ [] \ [] \\ & \text{list\_all2 } R (x21 \# x22) (y21 \# y22) = (R x21 y21 \wedge \text{list\_all2 } R \\ & \quad x22 y22) \end{aligned}$$

***t.rel\_distinct*** [*simp*]:

$$\begin{aligned} & \neg \text{list\_all2 } R \ [] (y21 \# y22) \\ & \neg \text{list\_all2 } R (y21 \# y22) \ [] \end{aligned}$$

***t.rel\_intros***:

$$\begin{aligned} & \text{list\_all2 } R \ [] \ [] \\ & \llbracket R x21 y21; \text{list\_all2 } R x22 y22 \rrbracket \Longrightarrow \text{list\_all2 } R (x21 \# x22) (y21 \\ & \quad \# y22) \end{aligned}$$

***t.rel\_cases*** [*consumes 1, case\_names t<sub>1</sub> ... t<sub>m</sub>, cases pred*]:

$$\begin{aligned} & \llbracket \text{list\_all2 } R a b; \llbracket a = []; b = [] \rrbracket \Longrightarrow \text{thesis}; \bigwedge x1 x2 y1 y2. \llbracket a = x1 \\ & \quad \# x2; b = y1 \# y2; R x1 y1; \text{list\_all2 } R x2 y2 \rrbracket \Longrightarrow \text{thesis} \rrbracket \Longrightarrow \\ & \quad \text{thesis} \end{aligned}$$

***t.rel\_sel***:

$$\begin{aligned} & \text{list\_all2 } R a b = (\text{null } a = \text{null } b \wedge (\neg \text{null } a \longrightarrow \neg \text{null } b \longrightarrow R \\ & \quad (\text{hd } a) (\text{hd } b) \wedge \text{list\_all2 } R (\text{tl } a) (\text{tl } b))) \end{aligned}$$

In addition, equational versions of *t.rel\_inject* and *rel\_distinct* are registered with the [*code*] attribute. The [*code*] attribute is set by the *code* plugin (Section 8.1).

The second subgroup consists of more abstract properties of the set functions, the map function, the predicator, and the relator:

***t.inj\_map***:

$$\text{inj } f \Longrightarrow \text{inj } (\text{map } f)$$

***t.inj\_map\_strong***:

$$\begin{aligned} & \llbracket \bigwedge z za. \llbracket z \in \text{set } x; za \in \text{set } xa; f z = fa za \rrbracket \Longrightarrow z = za; \text{map } f x = \\ & \quad \text{map } fa xa \rrbracket \Longrightarrow x = xa \end{aligned}$$

***t.map\_comp***:

$$\text{map } g (\text{map } f v) = \text{map } (g \circ f) v$$

***t.map\_cong0***:

$$(\bigwedge z. z \in \text{set } x \Longrightarrow f z = g z) \Longrightarrow \text{map } f x = \text{map } g x$$

***t.map\_cong*** [*fundef\_cong*]:

$$\llbracket x = ya; \bigwedge z. z \in \text{set } ya \Longrightarrow f z = g z \rrbracket \Longrightarrow \text{map } f x = \text{map } g ya$$

***t.map\_cong\_pred***:

$$\llbracket x = ya; \text{list\_all } (\lambda z. f z = g z) ya \rrbracket \Longrightarrow \text{map } f x = \text{map } g ya$$

***t.map\_cong\_simp***:

$$\llbracket x = ya; \bigwedge z. z \in \text{set } ya = \text{simp} \Rightarrow f z = g z \rrbracket \Longrightarrow \text{map } f x = \text{map } g ya$$

***t.map\_id0***:

$$\text{map } id = id$$

***t.map\_id***:

$$\text{map } id t = t$$

***t.map\_ident***:

$$\text{map } (\lambda x. x) t = t$$

***t.map\_ident\_strong***:

$$(\bigwedge z. z \in \text{set } t \Longrightarrow f z = z) \Longrightarrow \text{map } f t = t$$

***t.map\_transfer*** [*transfer\_rule*]:

$$\text{rel\_fun } (\text{rel\_fun } Rb Sd) (\text{rel\_fun } (\text{list\_all2 } Rb) (\text{list\_all2 } Sd)) \text{ map } \text{map}$$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

***t.pred\_cong*** [*fundef\_cong*]:

$$\llbracket x = ya; \bigwedge z. z \in \text{set } ya \Longrightarrow P z = Pa z \rrbracket \Longrightarrow \text{list\_all } P x = \text{list\_all } Pa ya$$

***t.pred\_cong\_simp***:

$$\llbracket x = ya; \bigwedge z. z \in \text{set } ya = \text{simp} \Rightarrow P z = Pa z \rrbracket \Longrightarrow \text{list\_all } P x = \text{list\_all } Pa ya$$

***t.pred\_map***:

$$\text{list\_all } Q (\text{map } f x) = \text{list\_all } (Q \circ f) x$$

***t.pred\_mono*** [*mono*]:

$$P \leq Pa \Longrightarrow \text{list\_all } P \leq \text{list\_all } Pa$$

***t.pred\_mono\_strong***:

$$\llbracket \text{list\_all } P x; \bigwedge z. \llbracket z \in \text{set } x; P z \rrbracket \Longrightarrow Pa z \rrbracket \Longrightarrow \text{list\_all } Pa x$$

***t.pred\_rel***:

$$\text{list\_all } P x = \text{list\_all2 } (\text{eq\_onp } P) x x$$

***t.pred\_set***:

$$\text{list\_all } P = (\lambda x. \text{Ball } (\text{set } x) P)$$

***t.pred\_transfer*** [*transfer\_rule*]:

$$\text{rel\_fun } (\text{rel\_fun } R (=)) (\text{rel\_fun } (\text{list\_all2 } R) (=)) \text{ list\_all } \text{list\_all}$$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

***t.pred\_True:***

$$\text{list\_all } (\lambda_. \text{ True}) = (\lambda_. \text{ True})$$

***t.set\_map:***

$$\text{set } (\text{map } f \ v) = f \ ' \ \text{set } v$$

***t.set\_transfer* [transfer\_rule]:**

$$\text{rel\_fun } (\text{list\_all2 } R) \ (\text{rel\_set } R) \ \text{set } \text{set}$$

The [transfer\_rule] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

***t.rel\_compp* [relator\_distr]:**

$$\text{list\_all2 } (R \ \text{OO } S) = \text{list\_all2 } R \ \text{OO } \text{list\_all2 } S$$

The [relator\_distr] attribute is set by the *lifting* plugin (Section 8.4).

***t.rel\_conversep:***

$$\text{list\_all2 } R^{-} = (\text{list\_all2 } R)^{-}$$

***t.rel\_eq:***

$$\text{list\_all2 } (=) = (=)$$

***t.rel\_eq\_onp:***

$$\text{list\_all2 } (\text{eq\_onp } P) = \text{eq\_onp } (\text{list\_all } P)$$

***t.rel\_flip:***

$$\text{list\_all2 } R^{-} \ a \ b = \text{list\_all2 } R \ b \ a$$

***t.rel\_map:***

$$\text{list\_all2 } S_b \ (\text{map } i \ x) \ y = \text{list\_all2 } (\lambda x. S_b \ (i \ x)) \ x \ y$$

$$\text{list\_all2 } S_a \ x \ (\text{map } g \ y) = \text{list\_all2 } (\lambda x \ y. S_a \ x \ (g \ y)) \ x \ y$$

***t.rel\_mono* [mono, relator\_mono]:**

$$R \leq Ra \implies \text{list\_all2 } R \leq \text{list\_all2 } Ra$$

The [relator\_mono] attribute is set by the *lifting* plugin (Section 8.4).

***t.rel\_mono\_strong:***

$$\begin{aligned} & \llbracket \text{list\_all2 } R \ x \ y; \bigwedge z \ yb. \llbracket z \in \text{set } x; \ yb \in \text{set } y; \ R \ z \ yb \rrbracket \implies Ra \ z \ yb \rrbracket \\ & \implies \text{list\_all2 } Ra \ x \ y \end{aligned}$$

***t.rel\_cong* [fundef\_cong]:**

$$\begin{aligned} & \llbracket x = ya; \ y = xa; \bigwedge z \ yb. \llbracket z \in \text{set } ya; \ yb \in \text{set } xa \rrbracket \implies R \ z \ yb = Ra \\ & \ z \ yb \rrbracket \implies \text{list\_all2 } R \ x \ y = \text{list\_all2 } Ra \ ya \ xa \end{aligned}$$

***t.rel\_cong\_simp:***

$$\begin{aligned} & \llbracket x = ya; \ y = xa; \bigwedge z \ yb. \ z \in \text{set } ya = \text{simp} \implies \ yb \in \text{set } xa = \text{simp} \implies \\ & \ R \ z \ yb = Ra \ z \ yb \rrbracket \implies \text{list\_all2 } R \ x \ y = \text{list\_all2 } Ra \ ya \ xa \end{aligned}$$

***t.rel\_refl:***

$$(\bigwedge x. Ra \ x \ x) \implies \text{list\_all2 } Ra \ x \ x$$

***t.rel\_refl\_strong***:

$$(\bigwedge z. z \in \text{set } x \implies Ra \ z \ z) \implies \text{list\_all2 } Ra \ x \ x$$

***t.rel\_refl***:

$$\text{reflp } R \implies \text{reflp } (\text{list\_all2 } R)$$

***t.rel\_symp***:

$$\text{symp } R \implies \text{symp } (\text{list\_all2 } R)$$

***t.rel\_transp***:

$$\text{transp } R \implies \text{transp } (\text{list\_all2 } R)$$

***t.rel\_transfer*** [*transfer\_rule*]:

$$\text{rel\_fun } (\text{rel\_fun } Sa \ (\text{rel\_fun } Sc \ (=))) \ (\text{rel\_fun } (\text{list\_all2 } Sa) \ (\text{rel\_fun } (\text{list\_all2 } Sc) \ (=))) \ \text{list\_all2 } \text{list\_all2}$$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

### 2.4.3 Inductive Theorems

The inductive theorems are as follows:

***t.induct*** [*case\_names*  $C_1 \dots C_n$ , *induct* *t*]:

$$\llbracket P \ \square; \bigwedge x1 \ x2. P \ x2 \implies P \ (x1 \ \# \ x2) \rrbracket \implies P \ \text{list}$$

***t.rel\_induct*** [*case\_names*  $C_1 \dots C_n$ , *induct* *pred*]:

$$\llbracket \text{list\_all2 } R \ x \ y; Q \ \square \ \square; \bigwedge a21 \ a22 \ b21 \ b22. \llbracket R \ a21 \ b21; Q \ a22 \ b22 \rrbracket \implies Q \ (a21 \ \# \ a22) \ (b21 \ \# \ b22) \rrbracket \implies Q \ x \ y$$

$t_1 \dots t_m$ .***induct*** [*case\_names*  $C_1 \dots C_n$ ]:

$t_1 \dots t_m$ .***rel\_induct*** [*case\_names*  $C_1 \dots C_n$ ]:

Given  $m > 1$  mutually recursive datatypes, this induction rule can be used to prove  $m$  properties simultaneously.

***t.rec*** [*simp*, *code*]:

$$\text{rec\_list } f1 \ f2 \ \square = f1$$

$$\text{rec\_list } f1 \ f2 \ (x21 \ \# \ x22) = f2 \ x21 \ x22 \ (\text{rec\_list } f1 \ f2 \ x22)$$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

***t.rec\_o\_map***:

$$\text{rec\_list } g \ ga \ \circ \ \text{map } f = \text{rec\_list } g \ (\lambda x \ xa. ga \ (f \ x) \ (\text{map } f \ xa))$$

***t.rec\_transfer*** [*transfer\_rule*]:

$$\text{rel\_fun } S \ (\text{rel\_fun } (\text{rel\_fun } R \ (\text{rel\_fun } (\text{list\_all2 } R) \ (\text{rel\_fun } S \ S))) \ (\text{rel\_fun } (\text{list\_all2 } R) \ S)) \ \text{rec\_list } \text{rec\_list}$$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

For convenience, `datatype` also provides the following collection:

```
t.simps = t.inject t.distinct t.case t.rec t.map t.rel_inject
          t.rel_distinct t.set
```

## 2.5 Proof Method

### 2.5.1 *countable\_datatype*

The theory `~/src/HOL/Library/Countable.thy` provides a proof method called *countable\_datatype* that can be used to prove the countability of many datatypes, building on the countability of the types appearing in their definitions and of any type arguments. For example:

```
instance list :: (countable) countable
by countable_datatype
```

## 2.6 Antiquotation

### 2.6.1 *datatype*

The *datatype* antiquotation, written `\<^datatype><t>` or `@{datatype t}`, where *t* is a type name, expands to L<sup>A</sup>T<sub>E</sub>X code for the definition of the datatype, with each constructor listed with its argument types. For example, if *t* is *option*:

```
datatype 'a option = None | Some 'a
```

## 2.7 Compatibility Issues

The command `datatype` has been designed to be highly compatible with the old, pre-Isabelle2015 command, to ease migration. There are nonetheless a few incompatibilities that may arise when porting:

- *The Standard ML interfaces are different.* Tools and extensions written to call the old ML interfaces will need to be adapted to the new interfaces. The *BNF\_LFP\_Compact* structure provides convenience functions that simulate the old interfaces in terms of the new ones.
- *The recursor `rec_t` has a different signature for nested recursive datatypes.* In the old package, nested recursion through non-functions was internally reduced to mutual recursion. This reduction was visible in

the type of the recursor, used by **primrec**. Recursion through functions was handled specially. In the new package, nested recursion (for functions and non-functions) is handled in a more modular fashion. The old-style recursor can be generated on demand using **primrec** if the recursion is via new-style datatypes, as explained in Section 3.1.5, or using **datatype\_compat**.

- *Accordingly, the induction rule is different for nested recursive datatypes.* Again, the old-style induction rule can be generated on demand using **primrec** if the recursion is via new-style datatypes, as explained in Section 3.1.5, or using **datatype\_compat**. For recursion through functions, the old-style induction rule can be obtained by applying the `[unfolded all_mem_range]` attribute on `t.induct`.
- *The size function has a slightly different definition.* The new function returns 1 instead of 0 for some nonrecursive constructors. This departure from the old behavior made it possible to implement `size` in terms of the generic function `t.size_t`. Moreover, the new function considers nested occurrences of a value, in the nested recursive case. The old behavior can be obtained by disabling the `size` plugin (Section 8) and instantiating the `size` type class manually.
- *The internal constructions are completely different.* Proof texts that unfold the definition of constants introduced by the old command will be difficult to port.
- *Some constants and theorems have different names.* For non-mutually recursive datatypes, the alias `t.inducts` for `t.induct` is no longer generated. For  $m > 1$  mutually recursive datatypes, `rec_t1...tm_i` has been renamed `rec_ti` for each  $i \in \{1, \dots, m\}$ , `t1...tm.inducts(i)` has been renamed `ti.induct` for each  $i \in \{1, \dots, m\}$ , and the collection `t1...tm.size` (generated by the `size` plugin, Section 8.2) has been divided into `t1.size`,  $\dots$ , `tm.size`.
- *The `t.simps` collection has been extended.* Previously available theorems are available at the same index as before.
- *Variables in generated properties have different names.* This is rarely an issue, except in proof texts that refer to variable names in the `[where ...]` attribute. The solution is to use the more robust `[of ...]` syntax.

### 3 Defining Primitively Recursive Functions

Recursive functions over datatypes can be specified using the **primrec** command, which supports primitive recursion, or using the **fun**, **function**, and **partial\_function** commands. In this tutorial, the focus is on **primrec**; **fun** and **function** are described in a separate tutorial [6].

Because it is restricted to primitive recursion, **primrec** is less powerful than **fun** and **function**. However, there are primitively recursive specifications (e.g., based on infinitely branching or mutually recursive datatypes) for which **fun**'s termination check fails. It is also good style to use the simpler **primrec** mechanism when it works, both as an optimization and as documentation.

#### 3.1 Introductory Examples

Primitive recursion is illustrated through concrete examples based on the datatypes defined in Section 2.1. More examples can be found in the directory `~/src/HOL/Datatype_Examples`.

##### 3.1.1 Nonrecursive Types

Primitive recursion removes one layer of constructors on the left-hand side in each equation. For example:

```
primrec (nonexhaustive) bool_of_tbool :: "tbool  $\Rightarrow$  bool" where
  "bool_of_tbool Faalse  $\longleftrightarrow$  False"
| "bool_of_tbool Truue  $\longleftrightarrow$  True"

primrec the_list :: "'a option  $\Rightarrow$  'a list" where
  "the_list None = []"
| "the_list (Some a) = [a]"

primrec the_default :: "'a  $\Rightarrow$  'a option  $\Rightarrow$  'a" where
  "the_default d None = d"
| "the_default _ (Some a) = a"

primrec mirrror :: "('a, 'b, 'c) triple  $\Rightarrow$  ('c, 'b, 'a) triple" where
  "mirrror (Triple a b c) = Triple c b a"
```

The equations can be specified in any order, and it is acceptable to leave out some cases, which are then unspecified. Pattern matching on the left-hand side is restricted to a single datatype, which must correspond to the same argument in all equations.



### 3.1.2 Simple Recursion

For simple recursive types, recursive calls on a constructor argument are allowed on the right-hand side:

```
primrec replicate :: "nat  $\Rightarrow$  'a  $\Rightarrow$  'a list" where
  "replicate Zero _ = []"
| "replicate (Succ n) x = x # replicate n x"

primrec (nonexhaustive) at :: "'a list  $\Rightarrow$  nat  $\Rightarrow$  'a" where
  "at (x # xs) j =
    (case j of
      Zero  $\Rightarrow$  x
    | Succ j'  $\Rightarrow$  at xs j)"

primrec tfold :: "('a  $\Rightarrow$  'b  $\Rightarrow$  'b)  $\Rightarrow$  ('a, 'b) tlist  $\Rightarrow$  'b" where
  "tfold _ (TNil y) = y"
| "tfold f (TCons x xs) = f x (tfold f xs)"
```

Pattern matching is only available for the argument on which the recursion takes place. Fortunately, it is easy to generate pattern-matching equations using the **simps\_of\_case** command provided by the theory `~/src/HOL/Library/Simps_Case_Conv.thy`.

```
simps_of_case at_simps_alt: at.simps
```

This generates the lemma collection `at_simps_alt`:

$$\text{at } (x \# xs) \text{ Zero} = x \quad \text{at } (xa \# xs) (\text{Succ } x) = \text{at } xs \ x$$

The next example is defined using **fun** to escape the syntactic restrictions imposed on primitively recursive functions:

```
fun at_least_two :: "nat  $\Rightarrow$  bool" where
  "at_least_two (Succ (Succ _))  $\longleftrightarrow$  True"
| "at_least_two _  $\longleftrightarrow$  False"
```

### 3.1.3 Mutual Recursion

The syntax for mutually recursive functions over mutually recursive data-types is straightforward:

```
primrec
  nat_of_even_nat :: "even_nat  $\Rightarrow$  nat" and
  nat_of_odd_nat :: "odd_nat  $\Rightarrow$  nat"
where
  "nat_of_even_nat Even_Zero = Zero"
| "nat_of_even_nat (Even_Succ n) = Succ (nat_of_odd_nat n)"
```

| “*nat\_of\_odd\_nat* (*Odd Succ n*) = *Succ (nat\_of\_even\_nat n)*”

**primrec**

*eval<sub>e</sub>* :: “(*'a* ⇒ *int*) ⇒ (*'b* ⇒ *int*) ⇒ (*'a, 'b*) *exp* ⇒ *int*” **and**

*eval<sub>t</sub>* :: “(*'a* ⇒ *int*) ⇒ (*'b* ⇒ *int*) ⇒ (*'a, 'b*) *trm* ⇒ *int*” **and**

*eval<sub>f</sub>* :: “(*'a* ⇒ *int*) ⇒ (*'b* ⇒ *int*) ⇒ (*'a, 'b*) *fct* ⇒ *int*”

**where**

“*eval<sub>e</sub>*  $\gamma$   $\xi$  (*Term t*) = *eval<sub>t</sub>*  $\gamma$   $\xi$  *t*”

| “*eval<sub>e</sub>*  $\gamma$   $\xi$  (*Sum t e*) = *eval<sub>t</sub>*  $\gamma$   $\xi$  *t* + *eval<sub>e</sub>*  $\gamma$   $\xi$  *e*”

| “*eval<sub>t</sub>*  $\gamma$   $\xi$  (*Factor f*) = *eval<sub>f</sub>*  $\gamma$   $\xi$  *f*”

| “*eval<sub>t</sub>*  $\gamma$   $\xi$  (*Prod f t*) = *eval<sub>f</sub>*  $\gamma$   $\xi$  *f* + *eval<sub>t</sub>*  $\gamma$   $\xi$  *t*”

| “*eval<sub>f</sub>*  $\gamma$   $\_$  (*Const a*) =  $\gamma$  *a*”

| “*eval<sub>f</sub>*  $\_$   $\xi$  (*Var b*) =  $\xi$  *b*”

| “*eval<sub>f</sub>*  $\gamma$   $\xi$  (*Expr e*) = *eval<sub>e</sub>*  $\gamma$   $\xi$  *e*”

Mutual recursion is possible within a single type, using **fun**:

**fun**

*even* :: “*nat* ⇒ *bool*” **and**

*odd* :: “*nat* ⇒ *bool*”

**where**

“*even Zero* = *True*”

| “*even (Succ n)* = *odd n*”

| “*odd Zero* = *False*”

| “*odd (Succ n)* = *even n*”

### 3.1.4 Nested Recursion

In a departure from the old datatype package, nested recursion is normally handled via the map functions of the nesting type constructors. For example, recursive calls are lifted to lists using *map*:

**primrec** *at<sub>ff</sub>* :: “*'a tree<sub>ff</sub>* ⇒ *nat list* ⇒ *'a*” **where**

“*at<sub>ff</sub>* (*Node<sub>ff</sub> a ts*) *js* =

(*case js* of

□ ⇒ *a*

| *j # js'* ⇒ *at (map (λt. at<sub>ff</sub> t *js'*) ts) j*”

The next example features recursion through the *option* type. Although *option* is not a new-style datatype, it is registered as a BNF with the map function *map\_option*:

**primrec** *sum\_btree* :: “(*'a::{zero,plus}*) *btree* ⇒ *'a*” **where**

“*sum\_btree (BNode a lt rt)* =

*a* + *the\_default* 0 (*map\_option sum\_btree lt*) +

*the\_default* 0 (*map\_option sum\_btree rt*)”

The same principle applies for arbitrary type constructors through which recursion is possible. Notably, the map function for the function type ( $\Rightarrow$ ) is simply composition ( $\circ$ ):

```
primrec relabel_ft :: "('a  $\Rightarrow$  'a)  $\Rightarrow$  'a ftree  $\Rightarrow$  'a ftree" where
  "relabel_ft f (FTLeaf x) = FTLeaf (f x)"
| "relabel_ft f (FTNode g) = FTNode (relabel_ft f  $\circ$  g)"
```

For convenience, recursion through functions can also be expressed using  $\lambda$ -abstractions and function application rather than through composition. For example:

```
primrec relabel_ft :: "('a  $\Rightarrow$  'a)  $\Rightarrow$  'a ftree  $\Rightarrow$  'a ftree" where
  "relabel_ft f (FTLeaf x) = FTLeaf (f x)"
| "relabel_ft f (FTNode g) = FTNode ( $\lambda$ x. relabel_ft f (g x))"
```

```
primrec (nonexhaustive) subtree_ft :: "'a  $\Rightarrow$  'a ftree  $\Rightarrow$  'a ftree" where
  "subtree_ft x (FTNode g) = g x"
```

For recursion through curried  $n$ -ary functions,  $n$  applications of ( $\circ$ ) are necessary. The examples below illustrate the case where  $n = 2$ :

```
datatype 'a ftree2 = FTLeaf2 'a | FTNode2 "'a  $\Rightarrow$  'a  $\Rightarrow$  'a ftree2"
```

```
primrec relabel_ft2 :: "('a  $\Rightarrow$  'a)  $\Rightarrow$  'a ftree2  $\Rightarrow$  'a ftree2" where
  "relabel_ft2 f (FTLeaf2 x) = FTLeaf2 (f x)"
| "relabel_ft2 f (FTNode2 g) = FTNode2 (( $\circ$ ) (( $\circ$ ) (relabel_ft2 f)) g)"
```

```
primrec relabel_ft2 :: "('a  $\Rightarrow$  'a)  $\Rightarrow$  'a ftree2  $\Rightarrow$  'a ftree2" where
  "relabel_ft2 f (FTLeaf2 x) = FTLeaf2 (f x)"
| "relabel_ft2 f (FTNode2 g) = FTNode2 ( $\lambda$ x y. relabel_ft2 f (g x y))"
```

```
primrec (nonexhaustive) subtree_ft2 :: "'a  $\Rightarrow$  'a  $\Rightarrow$  'a ftree2  $\Rightarrow$  'a ftree2" where
  "subtree_ft2 x y (FTNode2 g) = g x y"
```

For any datatype featuring nesting, the predicator can be used instead of the map function, typically when defining predicates. For example:

```
primrec increasing_tree :: "int  $\Rightarrow$  int treeff  $\Rightarrow$  bool" where
  "increasing_tree m (Nodeff n ts)  $\longleftrightarrow$ 
  n  $\geq$  m  $\wedge$  list_all (increasing_tree (n + 1)) ts"
```

### 3.1.5 Nested-as-Mutual Recursion

For compatibility with the old package, but also because it is sometimes convenient in its own right, it is possible to treat nested recursive datatypes as mutually recursive ones if the recursion takes place through new-style datatypes. For example:

**primrec** (*nonexhaustive*)  
 $at_{ff} :: 'a\ tree_{ff} \Rightarrow nat\ list \Rightarrow 'a$  **and**  
 $ats_{ff} :: 'a\ tree_{ff}\ list \Rightarrow nat \Rightarrow nat\ list \Rightarrow 'a$   
**where**  
 $at_{ff}\ (Node_{ff}\ a\ ts)\ js =$   
 $(case\ js\ of$   
 $\ [] \Rightarrow a$   
 $\ | j\ \#\ js' \Rightarrow ats_{ff}\ ts\ j\ js')$   
 $| at_{ff}\ (t\ \#\ ts)\ j =$   
 $(case\ j\ of$   
 $\ Zero \Rightarrow at_{ff}\ t$   
 $\ | Succ\ j' \Rightarrow ats_{ff}\ ts\ j')$

Appropriate induction rules are generated as  $at_{ff}.induct$ ,  $ats_{ff}.induct$ , and  $at_{ff\_ats_{ff}}.induct$ . The induction rules and the underlying recursors are generated dynamically and are kept in a cache to speed up subsequent definitions.


Here is a second example:

**primrec**  
 $sum\_btree :: ('a::\{zero,plus\})\ btree \Rightarrow 'a$  **and**  
 $sum\_btree\_option :: 'a\ btree\ option \Rightarrow 'a$   
**where**  
 $sum\_btree\ (BNode\ a\ lt\ rt) =$   
 $a + sum\_btree\_option\ lt + sum\_btree\_option\ rt$   
 $| sum\_btree\_option\ None = 0$   
 $| sum\_btree\_option\ (Some\ t) = sum\_btree\ t$


## 3.2 Command Syntax

### 3.2.1 primrec


**primrec** :  $local\_theory \rightarrow local\_theory$



*pr-options*



*pr-equation*



The **primrec** command introduces a set of mutually recursive functions over datatypes.

The syntactic entity *target* can be used to specify a local context, *fixes* denotes a list of names with optional type signatures, *thmdecl* denotes an optional name for the formula that follows, and *prop* denotes a HOL proposition [12].

The optional target is optionally followed by a combination of the following options:

- The *plugins* option indicates which plugins should be enabled (*only*) or disabled (*del*). By default, all plugins are enabled.
- The *nonexhaustive* option indicates that the functions are not necessarily specified for all constructors. It can be used to suppress the warning that is normally emitted when some constructors are missing.
- The *transfer* option indicates that an unconditional transfer rule should be generated and proved *by transfer\_prover*. The [*transfer\_rule*] attribute is set on the generated theorem.

### 3.3 Generated Theorems

The **primrec** command generates the following properties (listed for *tfold*):

*f.simps* [*simp*, *code*]:

*tfold uu (TNil y) = y*

*tfold f (TCons x xs) = f x (tfold f xs)*

The [*code*] attribute is set by the *code* plugin (Section 8.1).

*f.transfer* [*transfer\_rule*]:

*rel\_fun (rel\_fun R2 (rel\_fun R1 R1)) (rel\_fun (rel\_tlist R2 R1) R1) tfold tfold*

This theorem is generated by the *transfer* plugin (Section 8.3) for functions declared with the *transfer* option enabled.

*f.induct* [*case\_names C<sub>1</sub> ... C<sub>n</sub>*]:

This induction rule is generated for nested-as-mutual recursive functions (Section 3.1.5).

*f<sub>1</sub>...f<sub>m</sub>.induct* [*case\_names C<sub>1</sub> ... C<sub>n</sub>*]:

This induction rule is generated for nested-as-mutual recursive functions (Section 3.1.5). Given  $m > 1$  mutually recursive functions, this rule can be used to prove  $m$  properties simultaneously.

### 3.4 Recursive Default Values for Selectors

A datatype selector  $un\_D$  can have a default value for each constructor on which it is not otherwise specified. Occasionally, it is useful to have the default value be defined recursively. This leads to a chicken-and-egg situation, because the datatype is not introduced yet at the moment when the selectors are introduced. Of course, we can always define the selectors manually afterward, but we then have to state and prove all the characteristic theorems ourselves instead of letting the package do it.

Fortunately, there is a workaround that relies on overloading to relieve us from the tedium of manual derivations:

1. Introduce a fully unspecified constant  $un\_D_0 :: 'a$  using **consts**.
2. Define the datatype, specifying  $un\_D_0$  as the selector's default value.
3. Define the behavior of  $un\_D_0$  on values of the newly introduced datatype using the **overloading** command.
4. Derive the desired equation on  $un\_D$  from the characteristic equations for  $un\_D_0$ .

The following example illustrates this procedure:

```
consts termi0 :: 'a
```

```

datatype ('a, 'b) tlist =
  TNil (termi: 'b)
| TCons (thd: 'a) (ttl: "('a, 'b) tlist")
where
  "ttl (TNil y) = TNil y"
| "termi (TCons _ xs) = termi_0 xs"

overloading
  termi_0  $\equiv$  "termi_0 :: ('a, 'b) tlist  $\Rightarrow$  'b"
begin
primrec termi_0 :: "('a, 'b) tlist  $\Rightarrow$  'b" where
  "termi_0 (TNil y) = y"
| "termi_0 (TCons x xs) = termi_0 xs"
end

lemma termi_TCons[simp]: "termi (TCons x xs) = termi xs"
by (cases xs) auto

```

### 3.5 Compatibility Issues

The command **primrec**'s behavior on new-style datatypes has been designed to be highly compatible with that for old, pre-Isabelle2015 datatypes, to ease migration. There is nonetheless at least one incompatibility that may arise when porting to the new package:

- *Some theorems have different names.* For  $m > 1$  mutually recursive functions,  $f_1 \dots f_m$ .simps has been broken down into separate sub-collections  $f_i$ .simps.

## 4 Defining Codatatypes

Codatatypes can be specified using the **codatatype** command. The command is first illustrated through concrete examples featuring different flavors of corecursion. More examples can be found in the directory `~/src/HOL/Datatype_Examples`. The *Archive of Formal Proofs* also includes some useful codatatypes, notably for lazy lists [7].

### 4.1 Introductory Examples

#### 4.1.1 Simple Corecursion

Non-corecursive codatatypes coincide with the corresponding datatypes, so they are rarely used in practice. *Corecursive codatatypes* have the same

syntax as recursive datatypes, except for the command name. For example, here is the definition of lazy lists:

```
codatatype (lset: 'a) llist =
  lnull: LNil
| LCons (lhd: 'a) (ltl: "'a llist")
for
  map: lmap
  rel: llist_all2
  pred: llist_all
where
  "ltl LNil = LNil"
```

Lazy lists can be infinite, such as  $LCons\ 0\ (LCons\ 0\ (\dots))$  and  $LCons\ 0\ (LCons\ 1\ (LCons\ 2\ (\dots)))$ . Here is a related type, that of infinite streams:

```
codatatype (sset: 'a) stream =
  SCons (shd: 'a) (stl: "'a stream")
for
  map: smap
  rel: stream_all2
```

Another interesting type that can be defined as a codatatype is that of the extended natural numbers:

```
codatatype enat = EZero | ESucc enat
```

This type has exactly one infinite element,  $ESucc\ (ESucc\ (ESucc\ (\dots)))$ , that represents  $\infty$ . In addition, it has finite values of the form  $ESucc\ (\dots\ (ESucc\ EZero)\ \dots)$ .

Here is an example with many constructors:

```
codatatype 'a process =
  Fail
| Skip (cont: "'a process")
| Action (prefix: 'a) (cont: "'a process")
| Choice (left: "'a process") (right: "'a process")
```

Notice that the *cont* selector is associated with both *Skip* and *Action*.

#### 4.1.2 Mutual Corecursion

The example below introduces a pair of *mutually corecursive* types:

```
codatatype even_enat = Even_EZero | Even_ESucc odd_enat
and odd_enat = Odd_ESucc even_enat
```



### 4.1.3 Nested Corecursion

The next examples feature *nested corecursion*:

```
codatatype 'a treeii = Nodeii (lblii: 'a) (subii: "'a treeii llist")
```


```
codatatype 'a treeis = Nodeis (lblis: 'a) (subis: "'a treeis fset")
```

```
codatatype 'a sm = SM (accept: bool) (trans: "'a ⇒ 'a sm")
```

## 4.2 Command Syntax

### 4.2.1 codatatype

```
codatatype : local_theory → local_theory
```



Definitions of codatatypes have almost exactly the same syntax as for datatypes (Section 2.2). The *discs\_sels* option is superfluous because discriminators and selectors are always generated for codatatypes.

### 4.3 Generated Constants

Given a codatatype  $(a_1, \dots, a_m) t$  with  $m > 0$  live type variables and  $n$  constructors  $t.C_1, \dots, t.C_n$ , the same auxiliary constants are generated as for datatypes (Section 2.3), except that the recursor is replaced by a dual concept:

Corecursor:  $t.corec_t$

### 4.4 Generated Theorems

The characteristic theorems generated by **codatatype** are grouped in three broad categories:

- The *free constructor theorems* (Section 2.4.1) are properties of the constructors and destructors that can be derived for any freely generated type.

- The *functorial theorems* (Section 2.4.2) are properties of datatypes related to their BNF nature.
- The *coinductive theorems* (Section 4.4.1) are properties of datatypes related to their coinductive nature.

The first two categories are exactly as for datatypes.

#### 4.4.1 Coinductive Theorems

The coinductive theorems are listed below for 'a *l*list':

***t.coinduct*** [*consumes m, case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>, coinduct t*]:  

$$\llbracket R \text{ llist llist}'; \bigwedge \text{ llist llist}'. R \text{ llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge (\neg \text{lnull llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow \text{lh d llist} = \text{lh d llist}' \wedge R (\text{ltl llist}) (\text{ltl llist}')) \rrbracket \implies \text{llist} = \text{llist}'$$

***t.coinduct\_strong*** [*consumes m, case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>*]:  

$$\llbracket R \text{ llist llist}'; \bigwedge \text{ llist llist}'. R \text{ llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge (\neg \text{lnull llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow \text{lh d llist} = \text{lh d llist}' \wedge (R (\text{ltl llist}) (\text{ltl llist}') \vee \text{ltl llist} = \text{ltl llist}')) \rrbracket \implies \text{llist} = \text{llist}'$$

***t.rel\_coinduct*** [*consumes m, case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>, coinduct pred*]:  

$$\llbracket P x y; \bigwedge \text{ llist llist}'. P \text{ llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge (\neg \text{lnull llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow R (\text{lh d llist}) (\text{lh d llist}') \wedge P (\text{ltl llist}) (\text{ltl llist}')) \rrbracket \implies \text{llist\_all2 } R x y$$

***t<sub>1</sub>...<sub>m</sub>.coinduct*** [*case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>*]

***t<sub>1</sub>...<sub>m</sub>.coinduct\_strong*** [*case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>*]:

***t<sub>1</sub>...<sub>m</sub>.rel\_coinduct*** [*case\_names t<sub>1</sub> ... t<sub>m</sub>, case\_conclusion D<sub>1</sub> ... D<sub>n</sub>*]:

Given  $m > 1$  mutually corecursive codatatypes, these coinduction rules can be used to prove  $m$  properties simultaneously.

***t<sub>1</sub>...<sub>m</sub>.set\_induct*** [*case\_names C<sub>1</sub> ... C<sub>n</sub>, induct set: set<sub>j</sub>\_t<sub>1</sub>, ..., induct set: set<sub>j</sub>\_t<sub>m</sub>*]:  

$$\llbracket x \in \text{lset } a; \bigwedge z1 z2. P z1 (\text{LCons } z1 z2); \bigwedge z1 z2 xa. \llbracket xa \in \text{lset } z2; P xa z2 \rrbracket \implies P xa (\text{LCons } z1 z2) \rrbracket \implies P x a$$

If  $m = 1$ , the attribute [*consumes 1*] is generated as well.

*t.corec*:

$$p \ a \Longrightarrow \text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a = \text{LNil}$$

$$\neg \ p \ a \Longrightarrow \text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a = \text{LCons } (g21 \ a) \ (\text{if } q22 \ a \ \text{then } g221 \ a \ \text{else } \text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ (g222 \ a))$$

*t.corec\_code* [*code*]:

$$\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a = (\text{if } p \ a \ \text{then } \text{LNil} \ \text{else } \text{LCons } (g21 \ a) \ (\text{if } q22 \ a \ \text{then } g221 \ a \ \text{else } \text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ (g222 \ a)))$$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

*t.corec\_disc*:

$$p \ a \Longrightarrow \text{lnull } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a)$$

$$\neg \ p \ a \Longrightarrow \neg \ \text{lnull } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a)$$

*t.corec\_disc\_iff* [*simp*]:

$$\text{lnull } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a) = p \ a$$

$$(\neg \ \text{lnull } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a)) = (\neg \ p \ a)$$

*t.corec\_sel* [*simp*]:

$$\neg \ p \ a \Longrightarrow \text{lhd } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a) = g21 \ a$$

$$\neg \ p \ a \Longrightarrow \text{ltl } (\text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ a) = (\text{if } q22 \ a \ \text{then } g221 \ a \ \text{else } \text{corec\_llist } p \ g21 \ q22 \ g221 \ g222 \ (g222 \ a))$$

*t.map\_o\_corec*:

$$\text{lmap } f \circ \text{corec\_llist } g \ ga \ gb \ gc \ gd = \text{corec\_llist } g \ (f \circ ga) \ gb \ (\text{lmap } f \circ gc) \ gd$$

*t.corec\_transfer* [*transfer\_rule*]:

$$\text{rel\_fun } (\text{rel\_fun } S \ (=)) \ (\text{rel\_fun } (\text{rel\_fun } S \ R)) \ (\text{rel\_fun } (\text{rel\_fun } S \ (=)) \ (\text{rel\_fun } (\text{rel\_fun } S \ (\text{llist\_all2 } R)) \ (\text{rel\_fun } (\text{rel\_fun } S \ S) \ (\text{rel\_fun } S \ (\text{llist\_all2 } R)))))) \ \text{corec\_llist } \text{corec\_llist}$$

The [*transfer\_rule*] attribute is set by the *transfer* plugin (Section 8.3) for type constructors with no dead type arguments.

For convenience, **codatatype** also provides the following collection:

$$t.\text{simps} = t.\text{inject } t.\text{distinct } t.\text{case } t.\text{corec\_disc\_iff } t.\text{corec\_sel} \\ t.\text{map } t.\text{rel\_inject } t.\text{rel\_distinct } t.\text{set}$$

## 4.5 Antiquotation

### 4.5.1 *codatatype*

The *codatatype* antiquotation, written  $\backslash\langle^{\wedge}\text{codatatype}\rangle\langle t \rangle$  or  $\@{\text{codatatype } t}$ , where *t* is a type name, expands to L<sup>A</sup>T<sub>E</sub>X code for the definition of

the codatatype, with each constructor listed with its argument types. For example, if  $t$  is *llist*:

```
codatatype 'a llist = LNil | LCons 'a ('a llist)
```

## 5 Defining Primitively Corecursive Functions

Corecursive functions can be specified using the **primcorec** and **primcorecursive** commands, which support primitive corecursion. Other approaches include the more general **partial\_function** command, the **corec** and **corecursive** commands, and techniques based on domains and topologies [8]. In this tutorial, the focus is on **primcorec** and **primcorecursive**; **corec** and **corecursive** are described in a separate tutorial [3]. More examples can be found in the directories `~/src/HOL/Datatype_Examples` and `~/src/HOL/Corec_Examples`.

Whereas recursive functions consume datatypes one constructor at a time, corecursive functions construct codatypes one constructor at a time. Partly reflecting a lack of agreement among proponents of coalgebraic methods, Isabelle supports three competing syntaxes for specifying a function  $f$ :

- The *destructor view* specifies  $f$  by implications of the form

$$\dots \implies is\_C_j (f x_1 \dots x_n)$$

and equations of the form

$$un\_C_j i (f x_1 \dots x_n) = \dots$$

This style is popular in the coalgebraic literature.

- The *constructor view* specifies  $f$  by equations of the form

$$\dots \implies f x_1 \dots x_n = C_j \dots$$

This style is often more concise than the previous one.

- The *code view* specifies  $f$  by a single equation of the form

$$f x_1 \dots x_n = \dots$$

with restrictions on the format of the right-hand side. Lazy functional programming languages such as Haskell support a generalized version of this style.

All three styles are available as input syntax. Whichever syntax is chosen, characteristic theorems for all three styles are generated.

## 5.1 Introductory Examples

Primitive corecursion is illustrated through concrete examples based on the codatatypes defined in Section 4.1. More examples can be found in the directory `~/src/HOL/Datatype_Examples`. The code view is favored in the examples below. Sections 5.1.5 and 5.1.6 present the same examples expressed using the constructor and destructor views.

### 5.1.1 Simple Corecursion

Following the code view, corecursive calls are allowed on the right-hand side as long as they occur under a constructor, which itself appears either directly to the right of the equal sign or in a conditional expression:

```
primcorec literate :: "('a ⇒ 'a) ⇒ 'a ⇒ 'a llist" where
  "literate g x = LCons x (literate g (g x))"
```

```
primcorec siterate :: "('a ⇒ 'a) ⇒ 'a ⇒ 'a stream" where
  "siterate g x = SCons x (siterate g (g x))"
```

The constructor ensures that progress is made—i.e., the function is *productive*. The above functions compute the infinite lazy list or stream  $[x, g x, g (g x), \dots]$ . Productivity guarantees that prefixes  $[x, g x, g (g x), \dots, (g \overset{\sim}{\sim} k) x]$  of arbitrary finite length  $k$  can be computed by unfolding the code equation a finite number of times.

Corecursive functions construct codatatype values, but nothing prevents them from also consuming such values. The following function drops every second element in a stream:

```
primcorec every_snd :: "'a stream ⇒ 'a stream" where
  "every_snd s = SCons (shd s) (stl (stl s))"
```

Constructs such as *let-in*, *if-then-else*, and *case-of* may appear around constructors that guard corecursive calls:

```
primcorec lapp :: "'a llist ⇒ 'a llist ⇒ 'a llist" where
  "lapp xs ys =
    (case xs of
      LNil ⇒ ys
    | LCons x xs' ⇒ LCons x (lapp xs' ys))"
```

For technical reasons, *case-of* is only supported for case distinctions on (co)datatypes that provide discriminators and selectors.

Pattern matching is not supported by **primcorec**. Fortunately, it is easy to generate pattern-matching equations using the **simps\_of\_case** command provided by the theory `~/src/HOL/Library/Simps_Case_Conv.thy`.

**simps\_of\_case** *lapp\_simps*: *lapp.code*

This generates the lemma collection *lapp\_simps*:

$$\begin{aligned} \text{lapp } LNil \text{ } ys &= ys \\ \text{lapp } (LCons \text{ } xa \text{ } x) \text{ } ys &= LCons \text{ } xa \text{ } (\text{lapp } x \text{ } ys) \end{aligned}$$

Corecursion is useful to specify not only functions but also infinite objects:

**primcorec** *infty* :: *enat* **where**  
 “*infty* = *ESucc infty*”

The example below constructs a pseudorandom process value. It takes a stream of actions (*s*), a pseudorandom function generator (*f*), and a pseudorandom seed (*n*):

**primcorec**  
*random\_process* :: “*a stream* ⇒ (*int* ⇒ *int*) ⇒ *int* ⇒ ‘*a process*”  
**where**  
 “*random\_process s f n* =  
 (*if n mod 4 = 0 then*  
   *Fail*  
   *else if n mod 4 = 1 then*  
     *Skip (random\_process s f (f n))*  
   *else if n mod 4 = 2 then*  
     *Action (shd s) (random\_process (stl s) f (f n))*  
   *else*  
     *Choice (random\_process (every\_snd s) (f ∘ f) (f n))*  
       (*random\_process (every\_snd (stl s)) (f ∘ f) (f (f n))*))”

The main disadvantage of the code view is that the conditions are tested sequentially. This is visible in the generated theorems. The constructor and destructor views offer nonsequential alternatives.

### 5.1.2 Mutual Corecursion

The syntax for mutually corecursive functions over mutually corecursive data-types is unsurprising:

**primcorec**  
*even\_infty* :: *even\_enat* **and**  
*odd\_infty* :: *odd\_enat*  
**where**  
 “*even\_infty* = *Even\_ESucc odd\_infty*”  
 | “*odd\_infty* = *Odd\_ESucc even\_infty*”

### 5.1.3 Nested Corecursion

The next pair of examples generalize the *literate* and *siterate* functions (Section 5.1.3) to possibly infinite trees in which subnodes are organized either as a lazy list ( $tree_{ii}$ ) or as a finite set ( $tree_{is}$ ). They rely on the map functions of the nesting type constructors to lift the corecursive calls:

**primcorec**  $iterate_{ii} :: ('a \Rightarrow 'a\ llist) \Rightarrow 'a \Rightarrow 'a\ tree_{ii}$  **where**  
 $iterate_{ii}\ g\ x = Node_{ii}\ x\ (lmap\ (iterate_{ii}\ g)\ (g\ x))$

**primcorec**  $iterate_{is} :: ('a \Rightarrow 'a\ fset) \Rightarrow 'a \Rightarrow 'a\ tree_{is}$  **where**  
 $iterate_{is}\ g\ x = Node_{is}\ x\ (fimage\ (iterate_{is}\ g)\ (g\ x))$

Both examples follow the usual format for constructor arguments associated with nested recursive occurrences of the datatype. Consider  $iterate_{ii}$ . The term  $g\ x$  constructs an  $'a\ llist$  value, which is turned into an  $'a\ tree_{ii}\ llist$  value using  $lmap$ .

This format may sometimes feel artificial. The following function constructs a tree with a single, infinite branch from a stream:

**primcorec**  $tree_{ii\_of\_stream} :: 'a\ stream \Rightarrow 'a\ tree_{ii}$  **where**  
 $tree_{ii\_of\_stream}\ s =$   
 $Node_{ii}\ (shd\ s)\ (lmap\ tree_{ii\_of\_stream}\ (LCons\ (stl\ s)\ LNil))$

A more natural syntax, also supported by Isabelle, is to move corecursive calls under constructors:

**primcorec**  $tree_{ii\_of\_stream} :: 'a\ stream \Rightarrow 'a\ tree_{ii}$  **where**  
 $tree_{ii\_of\_stream}\ s =$   
 $Node_{ii}\ (shd\ s)\ (LCons\ (tree_{ii\_of\_stream}\ (stl\ s))\ LNil)$

The next example illustrates corecursion through functions, which is a bit special. Deterministic finite automata (DFAs) are traditionally defined as 5-tuples  $(Q, \Sigma, \delta, q_0, F)$ , where  $Q$  is a finite set of states,  $\Sigma$  is a finite alphabet,  $\delta$  is a transition function,  $q_0$  is an initial state, and  $F$  is a set of final states. The following function translates a DFA into a state machine:

**primcorec**  $sm\_of\_dfa :: ('q \Rightarrow 'a \Rightarrow 'q) \Rightarrow 'q\ set \Rightarrow 'q \Rightarrow 'a\ sm$  **where**  
 $sm\_of\_dfa\ \delta\ F\ q = SM\ (q \in F)\ (sm\_of\_dfa\ \delta\ F \circ \delta\ q)$

The map function for the function type  $(\Rightarrow)$  is composition  $((\circ))$ . For convenience, corecursion through functions can also be expressed using  $\lambda$ -abstractions and function application rather than through composition. For example:

**primcorec**  $sm\_of\_dfa :: ('q \Rightarrow 'a \Rightarrow 'q) \Rightarrow 'q\ set \Rightarrow 'q \Rightarrow 'a\ sm$  **where**  
 $sm\_of\_dfa\ \delta\ F\ q = SM\ (q \in F)\ (\lambda a.\ sm\_of\_dfa\ \delta\ F\ (\delta\ q\ a))$

**primcorec**  $empty\_sm :: 'a\ sm$  **where**

“*empty\_sm* = *SM False* ( $\lambda\_.$  *empty\_sm*)”

**primcorec** *not\_sm* :: “*'a sm*  $\Rightarrow$  *'a sm*” **where**

“*not\_sm M* = *SM* ( $\neg$  *accept M*) ( $\lambda a.$  *not\_sm* (*trans M a*))”

**primcorec** *or\_sm* :: “*'a sm*  $\Rightarrow$  *'a sm*  $\Rightarrow$  *'a sm*” **where**

“*or\_sm M N* =

*SM* (*accept M*  $\vee$  *accept N*) ( $\lambda a.$  *or\_sm* (*trans M a*) (*trans N a*))”

For recursion through curried  $n$ -ary functions,  $n$  applications of ( $\circ$ ) are necessary. The examples below illustrate the case where  $n = 2$ :

**codatatype** (*'a, 'b*) *sm2* =

*SM2* (*accept2: bool*) (*trans2: 'a*  $\Rightarrow$  *'b*  $\Rightarrow$  (*'a, 'b*) *sm2*)”

**primcorec**

*sm2\_of\_dfa* :: “(*'q*  $\Rightarrow$  *'a*  $\Rightarrow$  *'b*  $\Rightarrow$  *'q*)  $\Rightarrow$  *'q set*  $\Rightarrow$  *'q*  $\Rightarrow$  (*'a, 'b*) *sm2*”

**where**

“*sm2\_of\_dfa*  $\delta$  *F q* = *SM2* (*q*  $\in$  *F*) (( $\circ$ ) (( $\circ$ ) (*sm2\_of\_dfa*  $\delta$  *F*)) ( $\delta$  *q*))”

**primcorec**

*sm2\_of\_dfa* :: “(*'q*  $\Rightarrow$  *'a*  $\Rightarrow$  *'b*  $\Rightarrow$  *'q*)  $\Rightarrow$  *'q set*  $\Rightarrow$  *'q*  $\Rightarrow$  (*'a, 'b*) *sm2*”

**where**

“*sm2\_of\_dfa*  $\delta$  *F q* = *SM2* (*q*  $\in$  *F*) ( $\lambda a b.$  *sm2\_of\_dfa*  $\delta$  *F* ( $\delta$  *q a b*))”

#### 5.1.4 Nested-as-Mutual Corecursion

Just as it is possible to recurse over nested recursive datatypes as if they were mutually recursive (Section 3.1.5), it is possible to pretend that nested codatatypes are mutually corecursive. For example:

**primcorec**

*iterate<sub>ii</sub>* :: “(*'a*  $\Rightarrow$  *'a llist*)  $\Rightarrow$  *'a*  $\Rightarrow$  *'a tree<sub>ii</sub>*” **and**

*iterates<sub>ii</sub>* :: “(*'a*  $\Rightarrow$  *'a llist*)  $\Rightarrow$  *'a llist*  $\Rightarrow$  *'a tree<sub>ii</sub> llist*”

**where**

“*iterate<sub>ii</sub>* *g x* = *Node<sub>ii</sub>* *x* (*iterates<sub>ii</sub>* *g* (*g x*))”

| “*iterates<sub>ii</sub>* *g xs* =

(*case xs of*

*LNil*  $\Rightarrow$  *LNil*

| *LCons x xs'*  $\Rightarrow$  *LCons* (*iterate<sub>ii</sub>* *g x*) (*iterates<sub>ii</sub>* *g xs'*))”

Coinduction rules are generated as *iterate<sub>ii</sub>.coinduct*, *iterates<sub>ii</sub>.coinduct*, and *iterate<sub>ii</sub>\_iterates<sub>ii</sub>.coinduct* and analogously for *coinduct\_strong*. These rules and the underlying corecursors are generated dynamically and are kept in a cache to speed up subsequent definitions.



### 5.1.5 Constructor View

The constructor view is similar to the code view, but there is one separate conditional equation per constructor rather than a single unconditional equation. Examples that rely on a single constructor, such as *literate* and *siterate*, are identical in both styles.

Here is an example where there is a difference:

```
primcorec lapp :: "'a llist ⇒ 'a llist ⇒ 'a llist" where
  "lnull xs ⇒ lnull ys ⇒ lapp xs ys = LNil"
| "_ ⇒ lapp xs ys = LCons (lhd (if lnull xs then ys else xs))
  (if xs = LNil then ltl ys else lapp (ltl xs) ys)"
```

With the constructor view, we must distinguish between the *LNil* and the *LCons* case. The condition for *LCons* is left implicit, as the negation of that for *LNil*.

For this example, the constructor view is slightly more involved than the code equation. Recall the code view version presented in Section 5.1.1. The constructor view requires us to analyze the second argument (*ys*). The code equation generated from the constructor view also suffers from this.

In contrast, the next example is arguably more naturally expressed in the constructor view:

```
primcorec
  random_process :: "'a stream ⇒ (int ⇒ int) ⇒ int ⇒ 'a process"
where
  "n mod 4 = 0 ⇒ random_process s f n = Fail"
| "n mod 4 = 1 ⇒
  random_process s f n = Skip (random_process s f (f n))"
| "n mod 4 = 2 ⇒
  random_process s f n = Action (shd s) (random_process (stl s) f (f n))"
| "n mod 4 = 3 ⇒
  random_process s f n = Choice (random_process (every_snd s) f (f n))
  (random_process (every_snd (stl s)) f (f n))"
```

Since there is no sequentiality, we can apply the equation for *Choice* without having first to discharge  $n \bmod 4 \neq 0$ ,  $n \bmod 4 \neq 1$ , and  $n \bmod 4 \neq 2$ . The price to pay for this elegance is that we must discharge exclusiveness proof obligations, one for each pair of conditions ( $n \bmod 4 = i$ ,  $n \bmod 4 = j$ ) with  $i < j$ . If we prefer not to discharge any obligations, we can enable the *sequential* option. This pushes the problem to the users of the generated properties.

### 5.1.6 Destructor View

The destructor view is in many respects dual to the constructor view. Conditions determine which constructor to choose, and these conditions are interpreted sequentially or not depending on the *sequential* option. Consider the following examples:

```
primcorec literate :: “(‘a ⇒ ‘a) ⇒ ‘a ⇒ ‘a llist” where
  “¬ lnull (literate _ x)”
| “lhd (literate _ x) = x”
| “ltl (literate g x) = literate g (g x)”
```

```
primcorec siterate :: “(‘a ⇒ ‘a) ⇒ ‘a ⇒ ‘a stream” where
  “shd (siterate _ x) = x”
| “stl (siterate g x) = siterate g (g x)”
```

```
primcorec every_snd :: “‘a stream ⇒ ‘a stream” where
  “shd (every_snd s) = shd s”
| “stl (every_snd s) = stl (stl s)”
```

The first formula in the *local.literate* specification indicates which constructor to choose. For *local.siterate* and *local.every\_snd*, no such formula is necessary, since the type has only one constructor. The last two formulas are equations specifying the value of the result for the relevant selectors. Corecursive calls appear directly to the right of the equal sign. Their arguments are unrestricted.

The next example shows how to specify functions that rely on more than one constructor:

```
primcorec lapp :: “‘a llist ⇒ ‘a llist ⇒ ‘a llist” where
  “llnull xs ⇒ lnull ys ⇒ lnull (lapp xs ys)”
| “lhd (lapp xs ys) = lhd (if lnull xs then ys else xs)”
| “ltl (lapp xs ys) = (if xs = LNil then ltl ys else lapp (ltl xs) ys)”
```

For a codatatype with  $n$  constructors, it is sufficient to specify  $n - 1$  discriminator formulas. The command will then assume that the remaining constructor should be taken otherwise. This can be made explicit by adding

```
“_ ⇒ ¬ lnull (lapp xs ys)”
```

to the specification. The generated selector theorems are conditional.

The next example illustrates how to cope with selectors defined for several constructors:

```
primcorec
  random_process :: “‘a stream ⇒ (int ⇒ int) ⇒ int ⇒ ‘a process”
where
  “n mod 4 = 0 ⇒ random_process s f n = Fail”
```

```

| “n mod 4 = 1 ⇒ is_Skip (random_process s f n)”
| “n mod 4 = 2 ⇒ is_Action (random_process s f n)”
| “n mod 4 = 3 ⇒ is_Choice (random_process s f n)”
| “cont (random_process s f n) = random_process s f (f n)” of Skip
| “prefix (random_process s f n) = shd s”
| “cont (random_process s f n) = random_process (stl s) f (f n)” of Action
| “left (random_process s f n) = random_process (every_snd s) f (f n)”
| “right (random_process s f n) = random_process (every_snd (stl s)) f (f n)”

```

Using the *of* keyword, different equations are specified for *cont* depending on which constructor is selected.

Here are more examples to conclude:

### primcorec

```
even_infty :: even_enat and
```

```
odd_infty :: odd_enat
```

### where

```
“even_infty ≠ Even_EZero”
```

```
| “un_Even_ESucc even_infty = odd_infty”
```

```
| “un_Odd_ESucc odd_infty = even_infty”
```

### primcorec iterate<sub>ii</sub> :: “(‘a ⇒ ‘a llist) ⇒ ‘a ⇒ ‘a tree<sub>ii</sub>” where

```
“lblii (iterateii g x) = x”
```


```
| “subii (iterateii g x) = lmap (iterateii g) (g x)”
```



## 5.2 Command Syntax

### 5.2.1 primcorec and primcorecursive

```
primcorec : local_theory → local_theory
```

```
primcorecursive : local_theory → proof(prove)
```



*pcr-options**pcr-formula*

The **primcorec** and **primcorecursive** commands introduce a set of mutually corecursive functions over codatatypes.

The syntactic entity *target* can be used to specify a local context, *fixes* denotes a list of names with optional type signatures, *thmdecl* denotes an optional name for the formula that follows, and *prop* denotes a HOL proposition [12].

The optional target is optionally followed by a combination of the following options:

- The *plugins* option indicates which plugins should be enabled (*only*) or disabled (*del*). By default, all plugins are enabled.
- The *sequential* option indicates that the conditions in specifications expressed using the constructor or destructor view are to be interpreted sequentially.
- The *exhaustive* option indicates that the conditions in specifications expressed using the constructor or destructor view cover all possible cases. This generally gives rise to an additional proof obligation.
- The *transfer* option indicates that an unconditional transfer rule should be generated and proved *by transfer\_prover*. The [*transfer\_rule*] attribute is set on the generated theorem.

The **primcorec** command is an abbreviation for **primcorecursive** with *by auto?* to discharge any emerging proof obligations.

### 5.3 Generated Theorems

The **primcorec** and **primcorecursive** commands generate the following properties (listed for *literate*):

**f.code** [*code*]:

$\text{literate } g \ x = \text{LCons } x \ (\text{literate } g \ (g \ x))$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

**f.ctr**:

$\text{literate } g \ x = \text{LCons } x \ (\text{literate } g \ (g \ x))$

**f.disc** [*simp*, *code*]:

$\neg \text{lnull } (\text{literate } g \ x)$

The [*code*] attribute is set by the *code* plugin (Section 8.1). The [*simp*] attribute is set only for functions for which *f.disc\_iff* is not available.

**f.disc\_iff** [*simp*]:

$\neg \text{lnull } (\text{literate } g \ x)$

This property is generated only for functions declared with the *exhaustive* option or whose conditions are trivially exhaustive.

**f.sel** [*simp*, *code*]:

$\neg \text{lnull } (\text{literate } g \ x)$

The [*code*] attribute is set by the *code* plugin (Section 8.1).

**f.exclude**:

These properties are missing for *literate* because no exclusiveness proof obligations arose. In general, the properties correspond to the discharged proof obligations.

**f.exhaust**:

This property is missing for *literate* because no exhaustiveness proof obligation arose. In general, the property correspond to the discharged proof obligation.

**f.coinduct** [*consumes* *m*, *case\_names* *t*<sub>1</sub> ... *t*<sub>*m*</sub>,  
*case\_conclusion* *D*<sub>1</sub> ... *D*<sub>*n*</sub>]:

This coinduction rule is generated for nested-as-mutual corecursive functions (Section 5.1.4).

$f.\mathbf{coinduct\_strong}$  [*consumes*  $m$ , *case\_names*  $t_1 \dots t_m$ ,  
*case\_conclusion*  $D_1 \dots D_n$ ]:

This coinduction rule is generated for nested-as-mutual corecursive functions (Section 5.1.4).

$f_1 \dots f_m.\mathbf{coinduct}$  [*case\_names*  $t_1 \dots t_m$ ,  
*case\_conclusion*  $D_1 \dots D_n$ ]:

This coinduction rule is generated for nested-as-mutual corecursive functions (Section 5.1.4). Given  $m > 1$  mutually corecursive functions, this rule can be used to prove  $m$  properties simultaneously.

$f_1 \dots f_m.\mathbf{coinduct\_strong}$  [*case\_names*  $t_1 \dots t_m$ ,  
*case\_conclusion*  $D_1 \dots D_n$ ]:

This coinduction rule is generated for nested-as-mutual corecursive functions (Section 5.1.4). Given  $m > 1$  mutually corecursive functions, this rule can be used to prove  $m$  properties simultaneously.

For convenience, **primcorec** and **primcorecursive** also provide the following collection:

$f.\mathbf{simps} = f.\mathbf{disc\_iff}$  (or  $f.\mathbf{disc}$ )  $t.sel$

## 6 Registering Bounded Natural Functors

The (co)datatype package can be set up to allow nested recursion through arbitrary type constructors, as long as they adhere to the BNF requirements and are registered as BNFs. It is also possible to declare a BNF abstractly without specifying its internal structure.

### 6.1 Bounded Natural Functors

Bounded natural functors (BNFs) are a semantic criterion for where (co)recursion may appear on the right-hand side of an equation [4, 11].

An  $n$ -ary BNF is a type constructor equipped with a map function (functorial action),  $n$  set functions (natural transformations), and an infinite cardinal bound that satisfy certain properties. For example,  $'a\ llist$  is a unary BNF. Its predicator  $llist\_all :: ('a \Rightarrow bool) \Rightarrow 'a\ llist \Rightarrow bool$  extends unary predicates over elements to unary predicates over lazy lists. Similarly, its relator  $llist\_all2 :: ('a \Rightarrow 'b \Rightarrow bool) \Rightarrow 'a\ llist \Rightarrow 'b\ llist \Rightarrow bool$  extends binary predicates over elements to binary predicates over parallel lazy lists. The

cardinal bound limits the number of elements returned by the set function; it may not depend on the cardinality of  $'a$ .

The type constructors introduced by **datatype** and **codatatype** are automatically registered as BNFs. In addition, a number of old-style datatypes and non-free types are preregistered.

Given an  $n$ -ary BNF, the  $n$  type variables associated with set functions, and on which the map function acts, are *live*; any other variables are *dead*. Nested (co)recursion can only take place through live variables.

## 6.2 Introductory Examples

The example below shows how to register a type as a BNF using the **bnf** command. Some of the proof obligations are best viewed with the bundle "cardinal\_syntax" included.

The type is simply a copy of the function space  $'d \Rightarrow 'a$ , where  $'a$  is live and  $'d$  is dead. We introduce it together with its map function, set function, predicator, and relator.

```

typedef ('d, 'a) fn = "UNIV :: ('d  $\Rightarrow$  'a) set"
  by simp

setup_lifting type_definition_fn

lift_definition map_fn :: "('a  $\Rightarrow$  'b)  $\Rightarrow$  ('d, 'a) fn  $\Rightarrow$  ('d, 'b) fn" is "( $\circ$ )" .
lift_definition set_fn :: "('d, 'a) fn  $\Rightarrow$  'a set" is range .

lift_definition
  pred_fn :: "('a  $\Rightarrow$  bool)  $\Rightarrow$  ('d, 'a) fn  $\Rightarrow$  bool"
is
  "pred_fun ( $\lambda$ _. True)" .

lift_definition
  rel_fn :: "('a  $\Rightarrow$  'b  $\Rightarrow$  bool)  $\Rightarrow$  ('d, 'a) fn  $\Rightarrow$  ('d, 'b) fn  $\Rightarrow$  bool"
is
  "rel_fun (=)" .

bnf "('d, 'a) fn"
  map: map_fn
  sets: set_fn
  bd: "natLeq +c card_suc |UNIV :: 'd set|"
  rel: rel_fn
  pred: pred_fn
proof -
  show "map_fn id = id"

```

```

    by transfer auto
  next
  fix f :: "'a ⇒ 'b" and g :: "'b ⇒ 'c"
  show "map_fn (g ∘ f) = map_fn g ∘ map_fn f"
    by transfer (auto simp add: comp_def)
  next
  fix F :: "('d, 'a) fn" and f g :: "'a ⇒ 'b"
  assume "∧x. x ∈ set_fn F ⇒ f x = g x"
  then show "map_fn f F = map_fn g F"
    by transfer auto
  next
  fix f :: "'a ⇒ 'b"
  show "set_fn ∘ map_fn f = (·) f ∘ set_fn"
    by transfer (auto simp add: comp_def)
  next
  show "card_order (natLeq +c card_suc |UNIV :: 'd set|)"
    by (rule card_order_bd_fun)
  next
  show "cinfinitesimal (natLeq +c card_suc |UNIV :: 'd set|)"
    by (rule Cinfinitesimal_bd_fun[THEN conjunct1])
  next
  show "regularCard (natLeq +c card_suc |UNIV :: 'd set|)"
    by (rule regularCard_bd_fun)
  next
  fix F :: "('d, 'a) fn"
  have "|set_fn F| ≤o |UNIV :: 'd set|" (is "_ ≤o ?U")
    by transfer (rule card_of_image)
  also have "?U <o card_suc ?U"
    by (simp add: card_of_card_order_on card_suc_greater)
  also have "card_suc ?U ≤o natLeq +c card_suc ?U"
    using Card_order_card_suc card_of_card_order_on ordLeq_csum2 by
blast
  finally show "|set_fn F| <o natLeq +c card_suc |UNIV :: 'd set|".
  next
  fix R :: "'a ⇒ 'b ⇒ bool" and S :: "'b ⇒ 'c ⇒ bool"
  show "rel_fn R OO rel_fn S ≤ rel_fn (R OO S)"
    by (rule, transfer) (auto simp add: rel_fun_def)
  next
  fix R :: "'a ⇒ 'b ⇒ bool"
  show "rel_fn R = (λx y. ∃z. set_fn z ⊆ {(x, y). R x y} ∧ map_fn fst z =
x ∧ map_fn snd z = y)"
    unfolding fun_eq_iff relcompp.simps conversep.simps
    by transfer (force simp: rel_fun_def subset_iff)

```



```

next
  fix  $P :: 'a \Rightarrow bool$ 
  show “ $pred\_fn\ P = (\lambda x. Ball\ (set\_fn\ x)\ P)$ ”
  unfolding  $fun\_eq\_iff$  by  $transfer\ simp$ 
qed

print_theorems
print_bnfs

```

Using `print_theorems` and `print_bnfs`, we can contemplate and show the world what we have achieved.

This particular example does not need any nonemptiness witness, because the one generated by default is good enough, but in general this would be necessary. See `~/src/HOL/Basic_BNFs.thy`, `~/src/HOL/Library/Countable_Set_Type.thy`, `~/src/HOL/Library/FSet.thy`, and `~/src/HOL/Library/Multiset.thy` for further examples of BNF registration, some of which feature custom witnesses.

For many typedefs and quotient types, lifting the BNF structure from the raw type to the abstract type can be done uniformly. This is the task of the `lift_bnf` command. Using `lift_bnf`, the above registration of  $(d, a)\ fn$  as a BNF becomes much shorter:

```

lift_bnf  $(d, a)\ fn$ 
by  $force+$ 

```

For type copies (typedefs with `UNIV` as the representing set), the proof obligations are so simple that they can be discharged automatically, yielding another command, `copy_bnf`, which does not emit any proof obligations:

```

copy_bnf  $(d, a)\ fn$ 

```

Since record schemas are type copies, `copy_bnf` can be used to register them as BNFs:

```

record  $'a\ point =$ 
   $xval :: 'a$ 
   $yval :: 'a$ 

copy_bnf  $(a, z)\ point\_ext$ 

```

In the general case, the proof obligations generated by `lift_bnf` are simpler than the actual BNF properties. In particular, no cardinality reasoning is required. Consider the following type of nonempty lists:

```

typedef  $'a\ nonempty\_list = \{\!xs :: 'a\ list.\ xs \neq []\}$  by  $auto$ 

```

The `lift_bnf` command requires us to prove that the set of nonempty lists is closed under the map function and the zip function. The latter only occurs implicitly in the goal, in form of the variable `zs`.

```

lift_bnf 'a nonempty_list
proof -
  fix f and xs :: "'a list"
  assume "xs ∈ {xs. xs ≠ []}"
  then show "map f xs ∈ {xs. xs ≠ []}"
    by (cases xs) auto
next
  fix zs :: "('a × 'b) list"
  assume "map fst zs ∈ {xs. xs ≠ []}" "map snd zs ∈ {xs. xs ≠ []}"
  then show "∃ zs' ∈ {xs. xs ≠ []}.
    set zs' ⊆ set zs ∧
    map fst zs' = map fst zs ∧
    map snd zs' = map snd zs"
    by (cases zs) (auto intro!: exI[of _ zs])
qed

```

The `lift_bnf` command also supports quotient types. Here is an example that defines the option type as a quotient of the sum type. The proof obligations generated by `lift_bnf` for quotients are different from the ones for typedefs. You can find additional examples of usages of `lift_bnf` for both quotients and subtypes in the session *HOL-Datatype\_Examples*.

```

inductive ignore_Inl :: "'a + 'a ⇒ 'a + 'a ⇒ bool" where
  "ignore_Inl (Inl x) (Inl y)"
| "ignore_Inl (Inr x) (Inr x)"

```

```

lemma ignore_Inl_equivp:
  "ignore_Inl x x"
  "ignore_Inl x y ⇒ ignore_Inl y x"
  "ignore_Inl x y ⇒ ignore_Inl y z ⇒ ignore_Inl x z"
by (cases x; cases y; cases z; auto)+

```

```

quotient_type 'a myoption = "'a + 'a" / ignore_Inl
unfolding equivp_refl_p_symp_transp_refl_def symp_def transp_def
by (blast intro: ignore_Inl_equivp)

```

```

lift_bnf 'a myoption
proof -
  fix P :: "'a ⇒ 'b ⇒ bool" and Q :: "'b ⇒ 'c ⇒ bool"
  assume "P OO Q ≠ bot"
  then show "rel_sum P P OO ignore_Inl OO rel_sum Q Q
    ≤ ignore_Inl OO rel_sum (P OO Q) (P OO Q) OO ignore_Inl"
    by (fastforce)

```

```

next
  fix  $S :: \text{"'a set set"}$ 
  let  $?eq = \text{"\{(x, x^\wedge). ignore\_Inl x x^\wedge\}"}$ 
  let  $?in = \text{"\lambda A. \{x. Basic\_BNFs.setl x \cup Basic\_BNFs.setr x \subseteq A\}"}$ 
  assume  $\text{"S \neq \{\}"}$   $\text{"\bigcap S \neq \{\}"}$ 
  show  $\text{"(\bigcap A \in S. ?eq \text{" ?in A} \subseteq ?eq \text{" ?in (\bigcap S)"}$ 
  proof (intro subsetI)
    fix  $x$ 
    assume  $\text{"x \in (\bigcap A \in S. ?eq \text{" ?in A)"}$ 
    with  $\langle \bigcap S \neq \{\} \rangle$  show  $\text{"x \in ?eq \text{" ?in (\bigcap S)"}$ 
    by (cases x) (fastforce)+
  qed
qed

```

The next example declares a BNF axiomatically. This can be convenient for reasoning abstractly about an arbitrary BNF. The `bnf_axiomatization` command below introduces a type  $(a, b, c) F$ , three set constants, a map function, a predicator, a relator, and a nonemptiness witness that depends only on  $a$ . The type  $a \Rightarrow (a, b, c) F$  of the witness can be read as an implication: Given a witness for  $a$ , we can construct a witness for  $(a, b, c) F$ . The BNF properties are postulated as axioms.

```

bnf_axiomatization (setA: 'a, setB: 'b, setC: 'c)  $F$ 
  [wits: "a \Rightarrow (a, b, c) F"]


print_theorems
print_bnfs

```

## 6.3 Command Syntax

### 6.3.1 bnf

`bnf` : *local\_theory*  $\rightarrow$  *proof(prove)*




The `bnf` command registers an existing type as a bounded natural functor (BNF). The type must be equipped with an appropriate map function (functorial action). In addition, custom set functions, predicates, relators, and nonemptiness witnesses can be specified; otherwise, default versions are used.

The syntactic entity `target` can be used to specify a local context, `type` denotes a HOL type, and `term` denotes a HOL term [12].


The `plugins` option indicates which plugins should be enabled (*only*) or disabled (*del*). By default, all plugins are enabled.

### 6.3.2 lift\_bnf


$$\mathbf{lift\_bnf} : local\_theory \rightarrow proof(prove)$$



*lb-options*



*wit-terms*




The `lift_bnf` command registers as a BNF an existing type (the *abstract type*) that was defined as a subtype of a BNF (the *raw type*) using the `typedef` command or as a quotient type of a BNF (also, the *raw type*) using the `quotient_type`. To achieve this, it lifts the BNF structure on the raw type to the abstract type following a *type\_definition* or a *Quotient* theorem. The theorem is usually inferred from the type, but can also be explicitly supplied by means of the optional *via* clause. In case of quotients, it is sometimes also necessary to supply a second theorem of the form *reflp eq*, that expresses

the reflexivity (and thus totality) of the equivalence relation. In addition, custom names for the set functions, the map function, the predicator, and the relator, as well as nonemptiness witnesses can be specified.


Nonemptiness witnesses are not lifted from the raw type's BNF, as this would be incomplete. They must be given as terms (on the raw type) and proved to be witnesses. The command warns about witness types that are present in the raw type's BNF but not supplied by the user. The warning can be disabled by specifying the `no_warn_wits` option.

### 6.3.3 `copy_bnf`

`copy_bnf` :  $local\_theory \rightarrow local\_theory$




*cb-options*




The `copy_bnf` command performs the same lifting as `lift_bnf` for type copies (`typedefs` with `UNIV` as the representing set), without requiring the user to discharge any proof obligations or provide nonemptiness witnesses.

### 6.3.4 `bnf_axiomatization`

`bnf_axiomatization` :  $local\_theory \rightarrow local\_theory$



*wit-types*



The **bnf\_axiomatization** command declares a new type and associated constants (map, set, predicator, relator, and cardinal bound) and asserts the BNF properties for these constants as axioms.


The syntactic entity *target* can be used to specify a local context, *name* denotes an identifier, *typefree* denotes fixed type variable (*'a*, *'b*, ...), *mixfix* denotes the usual parenthesized mixfix notation, and *types* denotes a space-separated list of types [12].

The *plugins* option indicates which plugins should be enabled (*only*) or disabled (*del*). By default, all plugins are enabled.

Type arguments are live by default; they can be marked as dead by entering *dead* in front of the type variable (e.g., (*dead 'a*)) instead of an identifier for the corresponding set function. Witnesses can be specified by their types. Otherwise, the syntax of **bnf\_axiomatization** is identical to the left-hand side of a **datatype** or **codatatype** definition.

The command is useful to reason abstractly about BNFs. The axioms are safe because there exist BNFs of arbitrary large arities. Applications must import the `~/src/HOL/Library/BNF_Axiomatization.thy` theory to use this functionality.

6.3.5 `print_bnfs`


$$\text{print\_bnfs} : \text{local\_theory} \rightarrow$$


## 7 Deriving Destructors and Constructor Theorems

The derivation of convenience theorems for types equipped with free constructors, as performed internally by `datatype` and `codatatype`, is available as a stand-alone command called `free_constructors`.


### 7.1 Command Syntax

#### 7.1.1 `free_constructors`

$$\text{free\_constructors} : \text{local\_theory} \rightarrow \text{proof}(\text{prove})$$




*fc-ctor*



The **free\_constructors** command generates destructor constants for freely constructed types as well as properties about constructors and destructors. It also registers the constants and theorems in a data structure that is queried by various tools (e.g., **function**).


The syntactic entity *target* can be used to specify a local context, *name* denotes an identifier, *prop* denotes a HOL proposition, and *term* denotes a HOL term [12].

The syntax resembles that of **datatype** and **codatatype** definitions (Sections 2.2 and 4.2). A constructor is specified by an optional name for the discriminator, the constructor itself (as a term), and a list of optional names for the selectors.

Section 2.4 lists the generated theorems. For bootstrapping reasons, the generally useful [*fundef\_cong*] attribute is not set on the generated *case\_cong* theorem. It can be added manually using **declare**.

### 7.1.2 *simps\_of\_case*

**simps\_of\_case** : *local\_theory*  $\rightarrow$  *local\_theory*



The `simps_of_case` command provided by theory `~/src/HOL/Library/Simps_Case_Conv.thy` converts a single equation with a complex case expression on the right-hand side into a set of pattern-matching equations. For example,


`simps_of_case lapp_simps: lapp.code`

translates  $lapp\ xs\ ys = (case\ xs\ of\ LNil \Rightarrow ys \mid LCons\ x\ xs' \Rightarrow LCons\ x\ (lapp\ xs'\ ys))$  into

$$\begin{aligned} lapp\ LNil\ ys &= ys \\ lapp\ (LCons\ xa\ x)\ ys &= LCons\ xa\ (lapp\ x\ ys) \end{aligned}$$

### 7.1.3 case\_of\_simps

`case_of_simps : local_theory  $\rightarrow$  local_theory`



The `case_of_simps` command provided by theory `~/src/HOL/Library/Simps_Case_Conv.thy` converts a set of pattern-matching equations into single equation with a complex case expression on the right-hand side (cf. `simps_of_case`). For example,

`case_of_simps lapp_case: lapp_simps`

translates

$$\begin{aligned} lapp\ LNil\ ys &= ys \\ lapp\ (LCons\ xa\ x)\ ys &= LCons\ xa\ (lapp\ x\ ys) \end{aligned}$$

into  $lapp\ xba\ x3a = (case\ xba\ of\ LNil \Rightarrow x3a \mid LCons\ x2ba\ x1ba \Rightarrow LCons\ x2ba\ (lapp\ x1ba\ x3a))$ .

## 8 Selecting Plugins

Plugins extend the (co)datatype package to interoperate with other Isabelle packages and tools, such as the code generator, Transfer, Lifting, and Quickcheck. They can be enabled or disabled individually using the *plugins* option to the commands **datatype**, **primrec**, **codatatype**, **primcorec**, **primcorecursive**, **bnf**, **bnf\_axiomatization**, and **free\_constructors**. For example:

```
datatype (plugins del: code "quickcheck") color = Red | Black
```

Beyond the standard plugins, the *Archive of Formal Proofs* includes a **derive** command that derives class instances of datatypes [10].

### 8.1 Code Generator

The *code* plugin registers freely generated types, including (co)datatypes, and (co)recursive functions for code generation. No distinction is made between datatypes and codatatypes. This means that for target languages with a strict evaluation strategy (e.g., Standard ML), programs that attempt to produce infinite codatatype values will not terminate.

For types, the plugin derives the following properties:

*t.eq.refl* [*code nbe*]:

$$\text{equal\_class.equal } x \ x \equiv \text{True}$$

*t.eq.simps* [*code*]:

$$\text{equal\_class.equal } [] \ (x21 \# \ x22) \equiv \text{False}$$

$$\text{equal\_class.equal } (x21 \# \ x22) \ [] \equiv \text{False}$$

$$\text{equal\_class.equal } (x21 \# \ x22) \ [] \equiv \text{False}$$

$$\text{equal\_class.equal } [] \ (x21 \# \ x22) \equiv \text{False}$$

$$\text{equal\_class.equal } (x21 \# \ x22) \ (y21 \# \ y22) \equiv x21 = y21 \wedge x22 = y22$$

$$\text{equal\_class.equal } [] \ [] \equiv \text{True}$$

In addition, the plugin sets the [*code*] attribute on a number of properties of freely generated types and of (co)recursive functions, as documented in Sections 2.4, 3.3, 4.4, and 5.3.

### 8.2 Size

For each datatype *t*, the *size* plugin generates a generic size function *t.size\_t* as well as a specific instance *size* :: *t* ⇒ *nat* belonging to the *size* type class.

The **fun** command relies on *size* to prove termination of recursive functions on datatypes.

The plugin derives the following properties:

*t.size* [*simp*, *code*]:

$$\text{size\_list } x \ [] = 0$$

$$\text{size\_list } x \ (x21 \# x22) = x \ x21 + \text{size\_list } x \ x22 + \text{Suc } 0$$

$$\text{size } [] = 0$$

$$\text{size } (x21 \# x22) = \text{size } x22 + \text{Suc } 0$$

*t.size\_gen*:

$$\text{size\_list } x \ [] = 0$$

$$\text{size\_list } x \ (x21 \# x22) = x \ x21 + \text{size\_list } x \ x22 + \text{Suc } 0$$

*t.size\_gen\_o\_map*:

$$\text{size\_list } f \circ \text{map } g = \text{size\_list } (f \circ g)$$

*t.size\_neq*:

This property is missing for *'a list*. If the *size* function always evaluates to a non-zero value, this theorem has the form  $\text{size } x \neq 0$ .

The *t.size* and *t.size\_t* functions generated for datatypes defined by nested recursion through a datatype *u* depend on *u.size\_u*.

If the recursion is through a non-datatype *u* with type arguments  $'a_1, \dots, 'a_m$ , by default *u* values are given a size of 0. This can be improved upon by registering a custom size function of type  $('a_1 \Rightarrow \text{nat}) \Rightarrow \dots \Rightarrow ('a_m \Rightarrow \text{nat}) \Rightarrow u \Rightarrow \text{nat}$  using the ML function `BNF_LFP_Size.register_size` or `BNF_LFP_Size.register_size_global`. See theory `~/src/HOL/Library/Multiset.thy` for an example.

### 8.3 Transfer

For each (co)datatype with live type arguments and each manually registered BNF, the **transfer** plugin generates a predicator *t.pred\_t* and properties that guide the Transfer tool.

For types with at least one live type argument and *no dead type arguments*, the plugin derives the following properties:

*t.Domainp\_rel* [*relator\_domain*]:

$$\text{Domainp } (\text{list\_all2 } R) = \text{list\_all } (\text{Domainp } R)$$

*t.left\_total\_rel* [*transfer\_rule*]:

$$\text{left\_total } R \Longrightarrow \text{left\_total } (\text{list\_all2 } R)$$

$t.\mathbf{left\_unique\_rel}$  [*transfer\_rule*]:  
 $left\_unique\ R \implies left\_unique\ (list\_all2\ R)$   
 $t.\mathbf{right\_total\_rel}$  [*transfer\_rule*]:  
 $right\_total\ R \implies right\_total\ (list\_all2\ R)$   
 $t.\mathbf{right\_unique\_rel}$  [*transfer\_rule*]:  
 $right\_unique\ R \implies right\_unique\ (list\_all2\ R)$   
 $t.\mathbf{bi\_total\_rel}$  [*transfer\_rule*]:  
 $bi\_total\ R \implies bi\_total\ (list\_all2\ R)$   
 $t.\mathbf{bi\_unique\_rel}$  [*transfer\_rule*]:  
 $bi\_unique\ R \implies bi\_unique\ (list\_all2\ R)$

For (co)datatypes with at least one live type argument, the plugin sets the [*transfer\_rule*] attribute on the following (co)datatypes properties: *t.case\_transfer*, *t.sel\_transfer*, *t.ctr\_transfer*, *t.disc\_transfer*, *t.rec\_transfer*, and *t.corec\_transfer*. For (co)datatypes that further have *no dead type arguments*, the plugin sets [*transfer\_rule*] on *t.set\_transfer*, *t.map\_transfer*, and *t.rel\_transfer*.

For **primrec**, **primcorec**, and **primcorecursive**, the plugin implements the generation of the *f.transfer* property, conditioned by the *transfer* option, and sets the [*transfer\_rule*] attribute on these.

## 8.4 Lifting

For each (co)datatype and each manually registered BNF with at least one live type argument *and no dead type arguments*, the **lifting** plugin generates properties and attributes that guide the Lifting tool.

The plugin derives the following property:

$t.\mathbf{Quotient}$  [*quot\_map*]:  
 $Quotient\ R\ Abs\ Rep\ T \implies Quotient\ (list\_all2\ R)\ (map\ Abs)\ (map\ Rep)\ (list\_all2\ T)$

In addition, the plugin sets the [*relator\_eq*] attribute on a variant of the *t.rel\_eq\_onp* property, the [*relator\_mono*] attribute on *t.rel\_mono*, and the [*relator\_distr*] attribute on *t.rel\_compp*.

## 8.5 Quickcheck

The integration of datatypes with Quickcheck is accomplished by the **quickcheck** plugin. It combines a number of subplugins that instantiate specific

type classes. The subplugins can be enabled or disabled individually. They are listed below:

*quickcheck\_random*  
*quickcheck\_exhaustive*  
*quickcheck\_bounded\_forall*  
*quickcheck\_full\_exhaustive*  
*quickcheck\_narrowing*

## 8.6 Program Extraction

The *extraction* plugin provides realizers for induction and case analysis, to enable program extraction from proofs involving datatypes. This functionality is only available with full proof objects, i.e., with the *HOL-Proofs* session.

## 9 Known Bugs and Limitations

This section lists the known bugs and limitations of the (co)datatype package at the time of this writing.

1. *Defining mutually (co)recursive (co)datatypes can be slow.* Fortunately, it is always possible to recast mutual specifications to nested ones, which are processed more efficiently.
2. *Locally fixed types and terms cannot be used in type specifications.* The limitation on types can be circumvented by adding type arguments to the local (co)datatypes to abstract over the locally fixed types.
3. *The **primcorec** command does not allow user-specified names and attributes next to the entered formulas.* The less convenient syntax, using the **lemmas** command, is available as an alternative.
4. *The **primcorec** command does not allow corecursion under case-of for datatypes that are defined without discriminators and selectors.*
5. *There is no way to use an overloaded constant from a syntactic type class, such as 0, as a constructor.*
6. *There is no way to register the same type as both a datatype and a co-datatype.* This affects types such as the extended natural numbers, for which both views would make sense (for a different set of constructors).

7. *The names of variables are often suboptimal in the properties generated by the package.*
8. *The compatibility layer sometimes produces induction principles with a slightly different ordering of the premises than the old package.*

## Acknowledgment

Tobias Nipkow and Makarius Wenzel encouraged us to implement the new (co)datatype package. Andreas Lochbihler provided lots of comments on earlier versions of the package, especially on the coinductive part. Brian Huffman suggested major simplifications to the internal constructions. Ondřej Kunčar implemented the *transfer* and *lifting* plugins. Christian Sternagel and René Thiemann ported the **derive** command from the *Archive of Formal Proofs* to the new datatypes. Gerwin Klein and Lars Noschinski implemented the **simps\_of\_case** and **case\_of\_simps** commands. Florian Haftmann, Christian Urban, and Makarius Wenzel provided general advice on Isabelle and package writing. Stefan Milius and Lutz Schröder found an elegant proof that eliminated one of the BNF proof obligations. Mamoun Filali-Amine, Gerwin Klein, Andreas Lochbihler, Tobias Nipkow, and Christian Sternagel suggested many textual improvements to this tutorial.

## References

- [1] S. Berghofer and M. Wenzel. Inductive datatypes in HOL — lessons learned in Formal-Logic Engineering. In Y. Bertot, G. Dowek, A. Hirschowitz, C. Paulin, and L. Theys, editors, *Theorem Proving in Higher Order Logics: TPHOLs '99*, volume 1690 of *Lecture Notes in Computer Science*. Springer-Verlag, 1999.
- [2] J. C. Blanchette, J. Hölzl, A. Lochbihler, L. Panny, A. Popescu, and D. Traytel. Truly modular (co)datatypes for Isabelle/HOL. In G. Klein and R. Gamboa, editors, *5th International Conference on Interactive Theorem Proving, ITP 2014*, volume 8558 of *Lecture Notes in Computer Science*, pages 93–110. Springer, 2014.
- [3] J. C. Blanchette, A. Lochbihler, A. Popescu, and D. Traytel. *Defining Nonprimitively Corecursive Functions in Isabelle/HOL*. <https://isabelle.in.tum.de/doc/corec.pdf>.

- [4] J. C. Blanchette, A. Popescu, and D. Traytel. Witnessing (co)datatypes. In J. Vitek, editor, *24th European Symposium on Programming, ESOP 2015*, volume 9032 of *LNCS*, pages 359–382. Springer, 2015.
- [5] M. J. C. Gordon, R. Milner, and C. P. Wadsworth. *Edinburgh LCF: A Mechanised Logic of Computation*, volume 78 of *Lecture Notes in Computer Science*. Springer, 1979.
- [6] A. Krauss. *Defining Recursive Functions in Isabelle/HOL*. <https://isabelle.in.tum.de/doc/functions.pdf>.
- [7] A. Lochbihler. Coinductive. In G. Klein, T. Nipkow, and L. C. Paulson, editors, *The Archive of Formal Proofs*. <https://isa-afp.org/entries/Coinductive.shtml>, Feb. 2010.
- [8] A. Lochbihler and J. Hölzl. Recursive functions on lazy lists via domains and topologies. In G. Klein and R. Gamboa, editors, *Interactive Theorem Proving — 5th International Conference, ITP 2014*, volume 8558 of *Lecture Notes in Computer Science*, pages 341–357. Springer, 2014.
- [9] L. Panny, J. C. Blanchette, and D. Traytel. Primitively (co)recursive definitions for Isabelle/HOL. In *Isabelle Workshop 2014*, 2014.
- [10] C. Sternagel and R. Thiemann. Deriving class instances for datatypes. In G. Klein, T. Nipkow, and L. C. Paulson, editors, *The Archive of Formal Proofs*. <https://isa-afp.org/entries/Deriving.shtml>, March 2015.
- [11] D. Traytel, A. Popescu, and J. C. Blanchette. Foundational, compositional (co)datatypes for higher-order logic—Category theory applied to theorem proving. In *27th Annual IEEE Symposium on Logic in Computer Science, LICS 2012*, pages 596–605. IEEE, 2012.
- [12] M. Wenzel. *The Isabelle/Isar Reference Manual*. <https://isabelle.in.tum.de/doc/isar-ref.pdf>.